1. Tcl

스크립트 언어인 Tcl의 문법을 이해한다. 그리고 프로그램 개발의 편의를 도모하기 위해 Tcl이 제공하는 여러 가지 기능들을 간단히 살펴 본다.

1.1. 기본 문법
Tcl/Tk를 사용하기 위해 기본적으로 알아야 할 문법들에 관해서 설명 한다. Tcl/Tk의 프로그래밍 스타일, 주석, 변수, 명령어, 수학식,  변수 치환, 명령어 그루핑 등에 대하여 알아 본다.

1.1.1. 한 줄에 한 명령어 씩
Tcl인터프리터는 한 줄에 한 명령어씩만 나오는 것으로 생각한다. 그리고 모든 명령어는 반드시 한 줄 안에 기술되어야 한다. 

그러나 프로그램을 하다 보면 한 줄에 여러 개의 명령어를 쓰는 것이 이해에 도움을 주는 경우가 있다. 이 때에는 “ ; ”를 사용하여 각 명령어들을 구분시켜 줌으로써 한 줄에 여러 개의 명령어를 사용할 수 있다.

또 명령어가 길어져 여러 줄에 걸쳐서 기술해야 할 경우에는 “ \ ” 기호를 줄의 맨 마지막에 붙여서 다음 줄이 계속 한 명령어에 해당한다는 것을 인터프리터에 알려 줄 수 있다. 이 때 주의해야 할 점은 “ \ ” 기호 다음에 스페이스나 탭과 같은 공백 문자가 와서는 절대 안된다는 점이다. 이 방법 외에도 “ { } ”를 사용해서 여러 줄에 걸쳐 명령어를 기술할 수 있다. 중괄호를 이용한 그루핑에 대한 자세한 설명은 1.1.9절에 나와 있다.

	% set a 5; puts “a = $a”

a = 5

% set \

a 5

5


1.1.2. 주석 (comment)

Tcl인터프리터는 줄의 시작에 “ # ” 기호가 나타나면 그 줄 전체를 comment로 간주하고 무시한다. 줄의 시작점이 아닌 곳에 “ # ” 기호가 나오면 주석으로 간주하지 않는다. 

줄의 중간에 주석을 넣기 위해서는 “;”를 이용하면 된다. 그리고 여러 줄에 걸친 주석을 쓰기 위해서는 “ \ ”를 이용하면 된다. 백슬래쉬의 이런 용도에 대한 자세한 설명은 1.1.7절을 참조하기 바란다.

	# This is one line comment

set a 5; # To use comment in the middle of a line, Use “ ; “

# You can use multiline comment \

like this with “\”

# comment는 한글로 해도 된다.

# ‘#’의 위치는 반드시 첫번째 컬럼에 와야 한다.

       # 지금과 같이 ‘#’이 첫번째 컬럼이 아닌 곳에 오면 문법 오류가 발생한다.

if { $x < 0 } {

      # 지금과 같이 중괄호 속에 존재하는 주석은 ‘#’이 첫번째 컬럼에 오지 않아도 된다.

}




주석을 사용할 때 특히 주의해야 할 것은 주석은 반드시 명령어가 올 자리에 와야 한다는 점이다. “ { } ”를 사용한 복잡한 제어 명령의 경우 이 원칙을 지키는 것이 매우 중요하다. switch문을 설명할 때 좀 더 자세히 설명하겠다.

1.1.3. 명령어
Tcl의 명령어를 사용하는 방법은 다음과 같다.

command argument1 argument2 argument3 ...

첫번째로 명령어의 이름이 나오고 그 다음에 명령어로 전해 줄 Argument들이 나열 된다. 명령어와 각 Argument들의 구분은 Space나 Tab과 같은 공백문자로 이루어 진다. Tcl 명령어에는 Tcl 인터프리터 자체에서 제공하는 Built-In 명령어와 사용자가 정의한 함수들이 포함된다. Tcl함수를 정의하고 사용하는 방법에 대해서는 “1.7

 REF _Ref430745151 \* MERGEFORMAT 함수”에서 자세히 설명할 것이다.

다음과 같은 Tcl명령이 있을 때 

set a 5

이를 분석해 보면 명령어는 “set”이고 첫번째 Argument는 “a”이며 두 번째 Argument는 “5”이다.

1.1.4. 시스템 명령어
Tcl/Tk인터프리터는 C Shell스크립트에서 할 수 있는 모든 기능을 기본적으로 제공한다. 따라서 Tcl/Tk인터프리터에서는 UNIX와 C Shell이 기본적으로 제공하는 모든 명령어를 사용할 수 있다. 그리고 UNIX prompt상에서 수행할 수 있는 모든 명령어는 Tcl/Tk인터프리터 상에서도 수행할 수 있으며 스크립트 내에서도 마찬가지이다. 

그러나, Tcl/Tk 인터프리터 프롬프트 상에서는 바로 시스템 명령어를 실행할 수 있지만 스크립트에서는 바로 실행할 수 없다. 시스템 명령어를 스크립트 안에서 실행하려면 exec 명령어를 사용해야 한다는 점에 유의하기 바란다.

이러한 내용은 Windows 95/NT나 매킨토시에서도 마찬가지이다. 하지만 OS에 따라서 제공하는 시스템 명령어의 이름과 옵션이 다르다. 따라서 여러 가지 OS에서 사용할 Tcl/Tk스크립트 내에서는 시스템 명령어를 사용해서는 안된다. 굳이 사용해야 한다면 tcl_platform(platform)이라는 변수의 값을 확인하여 OS의 종류를 알아낸 다음 OS에 맞게 적절한 명령어를 사용하도록 프로그램해야 한다.

인터프리터 상에서 시스템 명령어를 사용하는 예는 다음과 같다.

	% ls -al

total 25

drwxrwxrwx   2 wonnr         512 Aug 21 11:39 .

drwxrwxrwx   4 wonnr         512 Aug 21 11:40 ..

-rwxrwxrwx   1 wonnr         386 Aug 21 10:38 SRAMView.resource

-rw-rw-rw-   1 wonnr        5740 Aug 21 11:37 readme.txt

-rw-rw-rw-   1 wonnr         373 Aug 21 11:37 sample.srv

-rw-rw-rw-   1 wonnr         386 Aug 21 11:37 sramView.resource

-rwxrwxrwx   1 wonnr       13820 Aug 21 11:37 sramView.tcl

%                                                              


스크립트 안에서 시스템 명령어를 사용하는 예는 다음과 같다.

	#!/usr/local/bin/tclsh

exec ls -al




1.1.5. 변수
Tcl에서 변수 이름에 사용될 수 있는 글자는 alphabet과 숫자, 밑줄 문자( _ )만이 가능하다고 알고 있는 편이 좋다. 다른 기호도 변수 이름에 사용될 수 있으나 이렇게 할 경우 소스 코드를 이해하기 어렵게 만들고 예기치 않은 논리적 에러를 발생시킬 확률이 높아 별로 권장하고 싶지 않다. 그리고 변수 이름은 숫자로 시작되면 안된다.

변수에 값을 할당하기 위해서는 set 명령어를 사용 한다. C 언어와 같이 “ = ”를 사용하여 변수 값을 할당할 수 없다는 점에 유의하기 바란다.

set variableName ?variableValue?

set명령어는 주어진 변수에 값을 할당하고 할당된 값을 리턴한다. Set 명령어에 Argument로 변수 이름만 주면 그 변수의 값을 리턴한다. 

변수의 값을 참조하기 위해서는 변수 이름 앞에 “ $ ”를 붙이면 된다.

	% set a 5

5

% set a

5

% set b $a

5


경우에 따라서 변수의 내용에 evaluation가능한 형태가 저장되어 있어 이를 막을 방법이 필요할 경우가 있다. 이러한 대표적인 경우가 다음과 같이 변수의 내용에 정규 표현식이 포함한 경우이다. 다음은 정규표현식에 사용될 패턴을 변수에 저장하고 사용하는 방법을 보여 주고 있다. 정규 표현식에 관해서는 “1.6

 REF _Ref430452451 \* MERGEFORMAT 정규 표현식”에서 보다 자세히 설명할 것이다.

	% set a {[a-z0-9]bc}

[a-z0-9]bc

% set b $a

error

% set b ${a}

[a-z0-9]bc


Tcl에서 변수는 C 언어에서처럼 미리 선언되어야 할 필요는 없다. set명령어에 의해 변수 값이 할당되는 순간 변수가 생성된다. 사용하지 않는 변수를 없애기 위해서는 unset명령어를 사용하면 된다. unset명령어는 이미 존재하는 명령어에 한해서만 사용할 수 있다. 존재하지 않는 변수에 unset명령어를 사용하면 다음과 같은 에러가 발생한다. 규모가 그리 크지 않은 프로그램에서는 unset을 사용하여 변수를 없애지 않아도 큰 문제는 없다.

	% set a

5

% unset a
% set a

can't read "a": no such variable

% unset c

can't read "c": no such variable


변수가 존재하는지 않는지를 알아보기 위해서는 info명령어에 첫번째 Argument로 exists를 주고 두 번째 Argument로 존재 여부를 알아보기 위한 변수 이름을 주면 된다. 존재하는 경우 1이 리턴되고 존재하지 않는 경우 0이 리턴된다.

	% set aa 2

2

% info exists aa

1

% info exists bb

 0


Tcl변수는 “ ”안에서 자유롭게 치환되므로 다음과 같이 사용할 수 있다.

	% set a 2

2

% puts “a is $a”

a is 2




다음과 같이 변수에 글자를 바로 붙여 써야할 경우에는 인터프리터가 변수 이름을 잘못 해석하는 수가 있다. 다음의 예제에서 “1st”와 같이 출력하려고 하는 것이지만 실제로 실행해 보면 에러가 난다. 이는 인터프리터가 변수의 이름을 “a”라고 인식하지 못하고 “ast”라고 인식했기 때문이다.

	% set a 1

1

% puts “$ast”

error




이런 경우에는 다음과 같이 2가지 해결책이 있다. 하나는 { }를 사용하는 방법이고, 다른 하나는 “ \ ”를 사용하는 방법이다.

	% set a 1

1

% puts “${a}st”

1st

% puts “$a\st”

1st


Tcl 변수에는 미리 정해진 형(Type)이 없다.  할당되는 값에 따라 변수의 형은 자동으로 바뀐다. 다시 말해서 정수형, 실수형 등과 같은 변수형이 없고 변수에 할당되는 값에 따라 적절한 형으로 바뀐다는 의미이다.  다음 예제에서 변수 “a”의 다양한 변신을 음미해 보기 바란다.

	% set a 2

2

% set a hahaha

hahaha

% set a 1.2

1.2

% set a “This is string”

This is string

% set a { one two three }

one two three


1.1.6. 수학식(Math Expression)

Tcl에서 수학식은 expr명령어를 사용하여 계산할 수 있다. 수학식의 표현 방법은 C언어와 유사하다.

	% expr 2 + 3

5

% expr ( 3 + 3 ) / 2

3

% expr cos(0)

1.0

% set x 3

3

% expr ( $x < 5 )? 1 : 2

1


연산자의 종류와 우선 순위를 정리하면 표 1과 같다. 우선 순위는 위쪽으로 갈 수록 높고 밑쪽으로 갈 수록 낮다. 

	연산자
	의미

	-  ~  !
	음수 부호,  bitwise NOT,  logical NOT

	*  /  %
	곱하기, 나누기, 나머지

	+  -
	더하기, 빼기

	<<  >>
	왼쪽 Shift, 오른쪽 Shift

	<  >  <=  >=
	작다, 크다, 작거나 같다, 크거나 같다

	==  !=
	같다, 다르다

	&
	bitwise AND

	^
	bitwise XOR

	|
	bitwise OR

	&&
	logical AND

	||
	logical OR

	x ? y : z

	If x then y else z


표 1  연산자의 종류와 우선순위
Tcl에서 제공하는 수학 함수들을 표 2에 정리하였다.

	수학함수
	의미

	acos(x)
	arc-cosine

	asin(x)
	arc-sine

	atan(x)
	arc-tangent

	atan2(y, x)
	(r, th)극에 대한 사각형 (x, y), atan2는 th의 값을 계산 함

	ceil(x)
	x와 같거나 큰 정수 중 가장 작은 수

	cos(x)
	cosine

	cosh(x)
	hyper cosine

	exp(x)
	자연 대수의 x 승

	floor(x)
	x와 같거나 작은 정수 중 가장 큰 수

	fmod(x,y)
	실수의 나머지 연산

	hypot(x,y)
	sqrt(x*x + y*y)

	log(x)
	로그, 밑이 자연 대수

	log10(x)
	로그, 밑이 10

	pow(x,y)
	x의 y승

	sin(x)
	sine

	sinh(x)
	hyper sine

	sqrt(x)
	제곱근

	tan(x)
	tangent

	tanh(x)
	hyper tangent

	abs(x)
	절대값

	double(x)
	실수로 변환 함

	int(x)
	정수로 변환, 소수점 이하 버림

	round(x)
	정수로 변환, 소수점 이하 반올림


표 2 Tcl Built-in 수학 함수
( ) = { } 로.. (위에꺼는 오타당)-호진 주

1.1.7. 백슬래쉬 문자열( Backslash Sequence)

백슬래쉬 다음에 오는 글자는 원래 의미를 잃어버리고 다른 의미로 사용된다. 가장 유용한 경우는 Tcl에서 의미있는 연산자를 기호로 사용하는 경우이다. 다음에 그 예제가 있다.

	% puts "[test]"

child process exited abnormally

% puts “\[test\] “

[test]


그 외에 Tcl에서 정의한 백슬래쉬 문자열들이 있다. 정리하면 다음과 같다.

	백슬래쉬 

문자열
	의미

	\a
	Bell (0x7)

	\b
	Backspace (0x8)

	\f
	Form Feed (0xC)

	\n
	New Line (0xA)

	\r
	Carriage return (0xD)

	\t
	Tab (0x9)

	\v
	Vertical Tab (0xB)

	\<new line>
	개행 문자가 백슬래쉬 바로 다음에 오면 다음 라인을 한 줄로 합침.한 명령어나 주석을 여러 줄에 걸쳐 사용할 때 유용한 기능임.

	\\
	Backslash 자체를 기호로 사용하기 위해

	\0
	8진수        ( puts “\041” 을 수행하면 ‘!’가 찍힘)

	\x
	16진수       ( puts “\x61”을 수행하면 ‘a’가 찍힘)

	\$
	$

	\”
	”

	\{
	{

	\}
	}

	\[
	[

	\]
	]


표 3 백슬래쉬 문자열
1.1.8. 명령어의 nesting

명령어의 Argument로 다른 명령어의 리턴 값이 들어갈 경우와 수학식 내에서 다른 명령어의 결과를 이용해야 할 경우 “[ ]”를 써서 명령어를 nesting시킬 수 있다. Tcl인터프리터는 “[ ]”를 만나면 “[ ]”내부에 있는 내용을 Tcl명령어로 보고 이를 실행시킨 후 그 결과를 “[ ]”가 있던 자리에 대치시킨다.

	% set a [ expr 3 + 2 ]

5

% set a [ expr 4 + [ expr 2 + 2 ] ]

8

% puts “ 7 + 3 = [ expr 7 + 3 ] “

7 + 3 = 10


“[ ]”를 명령어 자리에 사용하여 함수의 결과로 리턴되는 값을 명령어처럼 이용할 수도 있다. 다음에 그 예제가 있다. 프로그램 작성에 실제로 사용되는 경우는 거의 없겠지만 다음의 예제와 같은 프로그램도 작성할 수 있다. 함수의 작성 방법은 “1.7

 REF _Ref430745151 \* MERGEFORMAT 함수”에서 보다 자세히 설명할 것이다.

	% proc findCommand { what } {

    if { $what == 1 } {

        return set

    }

    return puts

}

% [findCommand 1] a 3

3

% [findCommand 0] “Hello! World”

Hello! World


1.1.9. 그루핑(Grouping)과 변수 치환(Substitution)

Tcl에서 Argument의 구분은 공백문자에 의해서 구분되는데 명령어에 넘겨줘야 할 Argument에 공백이 있을 경우 어떻게 해야할까? 이를 위해서 제공되는 것이 중괄호와 따옴표이다. 중괄호와 따옴표의 차이점은 중괄호는 변수 치환을 허용하지 않는 반면 따옴표는 변수 치환을 허용한다는 점이다. 다음의 예제를 통하여 이것을 쉽게 확인할 수 있을 것이다. 아래 두 경우 모두 명령어 puts에 전달되는 Argument는 그루핑에 의해 하나 밖에 없음에 유의해야 한다.

	% set a 5

% puts { a = $a }

a = $a

% puts “ a = $a “

a = 5


중괄호와 따옴표는 그 사이에 오는 모든 문자를 그루핑 해 준다. 따라서 중괄호와 따옴표를 이용해 여러 줄에 걸친 명령어를 기술할 수도 있다. Tcl인터프리터는 여는 중괄호를 만나면 닫는 중괄호가 나올 때까지 라인이 끝나지 않은 것으로 간주하고 계속 읽어들인다. 따옴표의 경우도 마찬가지 이다.

	% set a {

2

}

2

% puts “Multiple line

demo

haha “

Multiple line

demo

haha


1.1.10. 띄어 쓰기
Tcl/Tk는 띄어 쓰기에 좀 민감한 편이다. 특히 명령어 그루핑에 사용되는 “ { } ”와 명령어 nesting에 사용되는 “ [ ] ”를 사용할 때는 기호들 사이에 반드시 띄어 쓰기를 한다고 기억해 두면 좋다. 특히 닫는 괄호와 여는 괄호를 붙여서 사용한 경우에는 에러가 발생한다.

문법적으로 오류가 없는데도 불구하고 오류가 발생할 경우에는 닫는 괄호와 여는 괄호가 붙어 있는 경우가 없는지 확인해 보기 바란다. 다음의 예들은 자주 발생하는 실수이므로 오류의 원인을 확인해 보기 바란다.

	# This code has error

if { $value >= 0 }{


	# This is valid code

if { $value >= 0 } {


그리고 여는 중괄호가 키워드와 붙여 써지지는 않았는지 확인해 보기 바란다.

	# This code has error

if{ $value >= 0 } {


	# This is valid code

if { $value >= 0 } {

	


다음과 같이 닫는 사각 괄호가 맨 끝의 Argument와 붙어 있는 경우 서브루틴으로 닫는 사각 괄호도 함께 넘어가는 수가 있다.

	# This code has potential logical error

puts [ MySubFunc arg1 arg2]


	# This is valid code

puts [ MySubFunc arg1 arg2 ]


1.2. 제어구조
Tcl에서 제어구조 문법은 프로그램의 흐름을 제어하는 기능을 한다. 다른 프로그래밍 언어나 쉘 스크립트 언어에 익숙하다면 그 의미를 쉽게 파악할 수 있을 것이므로 문법과 간단한 예제만으로 설명을 대신하기로 한다.

1.2.1. if then else 구문
if booleanExp ?then? if_body ?else else_body?

	if {$x == 0} {

    puts “Zero Value”

} else {

    puts “Nonzero Value”

}


여기서 중요한 점은 if_body를 시작하는 중괄호가 반드시 if와 같은 라인에 있어야 한다. 줄이 바뀌기 전에 닫히지 않은 중괄호를 두어 인터프리터로 하여금 if_body가 여러 줄에 걸쳐 있음을 알려 주기 위함이다. 중괄호를 if다음 라인에 쓰기 위해서는 다음과 같이 백슬래쉬를 이용하여 기술하여야 한다.

	if {$x == 0} \

{

    puts “Zero Value”

} \

else \

{

    puts “Nonzero Value”

}


이러한 규칙은 다른 모든 제어 구문에도 동일하게 적용 된다. 

else if 문은 다음과 같이 사용한다.

	if { $x == 0 } {

    puts “Zero Value”

} elseif { $x > 0 } {

    puts “Plus Value”

} else {

    puts “Nonzero Value”

}


&&는 양쪽 모두가 참일 때만 참이다. 따라서 왼쪽 수식을 연산한 결과 거짓이면 오른 쪽 수식은 연산할 필요도 없이 그 결과는 거짓이 된다. 따라서 Tcl 인터프리터는 &&의 경우 먼저 왼쪽 수식을 연산하여 그 결과가 거짓이면 오른쪽의 수식을 실행하지 않는다. 따라서 다음과 같이 0으로 나누어지지 않도록 프로그램할 수 있다.

	if { $x =! 0 && [ expr $y / $x ] < 4 } {

    puts “hahaha”

}


||의 경우에는 어느 하나라도 참인 경우 참이다. 따라서 왼쪽 수식을 연산한 결과가 참이면 오른 쪽 수식은 연산할 필요도 없이 그 결과는 참이 된다. 따라서 Tcl인터프리터는 ||의 경우 먼저 왼쪽 수식을 연산하여 그 결과가 참이면 오른쪽의 수식을 실행하지 않는다. 따라서 ||의 경우에는 연산이 보다 간단한 것을 왼쪽에 두고 연산이 보다 복잡한 쪽을 오른쪽에 두는 것이 좋다.

1.2.2. switch 구문
switch options value pat1 pat1Body ?pat2 pat2Body … default defBody?
switch문은 value의 여러 가지 값에 따라 그 처리를 달리하는 경우에 사용된다. switch문에 가능한 옵션들은 다음과 같다.

*  -exact : value의 값은 정확히 패턴과 일치해야 한다. Tcl인터프리터는 다른 옵션을 따로 지정하지 않으면 이 옵션이 지정되어 있는 것으로 간주한다.

*  -glob : glob스타일 패턴 매칭
을 사용한다.

*  -regexp : 정규 표현식을 이용한 패턴 매칭을 사용한다.

*  -- : 옵션의 끝을 나타낸다. value값이 “ – ”로 시작하는 경우 옵션으로 오인하지 않기 위해 필요하다.

	switch –exact -- $value {

    Apple { incr apple_count }

    Orange { incr orange_count }

    default { incr nothing_count }

}


열거된 어떠한 패턴에도 매칭이 일어나지 않는 모든 경우 수행하고 싶은 일이 있으면 default라는 키워드를 써서 위와 같이 표현할 수 있다. 그러나 default를 맨 마지막에 두지 않으면 default가 키워드로 해석되지 않고 value의 값이 default일 때 수행해야 할 일로 해석된다는 점에 유의해야 한다. 아래의 예에서 위의 default 문장은 value의 값이 default일 때 default_count를 증가시키라는 뜻이다. 반면에 밑의 default문장은 value의 값이 default도 아니고 Apple도 아닐때 nothng_count를 증가시키라는 뜻이다.

	switch –exact -- $value {

    default { incr default_count }

    Apple { incr apple_count }

    default { incr nothing_count }

}


여러 패턴에 걸쳐 같은 일을 수행하고 싶은 일이 있을 때는 다음과 같이 “-”를 써서 연결할 수 있다.

	switch –exact -- $value {

    Apple –

    Banana -

    Orange { incr fruit_count }

}


switch문을 사용할 때 특히 주의해야 할 점은 주석을 패턴이 위치할 곳에 사용하면 인터프리터가 주석을 패턴으로 인식한다는 점이다. 주석은 반드시 명령어가 올 자리에 와야 한다.

	switch –exact -- $value {

    # This is bad comment

    Orange { 

        # This is good comment 

        incr orange_count

    }

}


1.2.3. while 구문
while booleanExp body 

while 문은 booleanExp가 참인 동안 body를 반복해서 수행 한다.

	set j 0

while { $j < 10 } {

    incr j

}


tcl에서 1을 참, 0을 거짓으로 해석하므로 다음과 같이 무한 루프를 만들 수 있다.

	while { 1 } {

    # this is infinite loop

}


1.2.4. for 구문
for initial loopcondition increment body

for문은 loopcondition이 참인 동안 body를 수행한다. 그리고 body의 수행이 끝나면 increment루틴을 수행하고 loopcondition이 만족되는지 조사한다.

	for { set k 0 } { $k < 10 } { incr k 2 } {

    puts $k

}


1.2.5. foreach 구문
foreach loopVariable valueList body
foreach는 valueList가 가지고 있는 모든 리스트
 원소들을 하나씩 차례로 loopVariable에 할당하면서 반복적으로 body를 수행한다.

	foreach element { one two three four } {

    puts $element

}

set var [ list 1 2 3 4 ]

foreach element $var {

    puts $element

}

set haha { apple kiwi orange }

foreach fruit $haha {

    puts $fruit

}




loopVariable에 여러 개의 변수가 올 수도 있다. 변수와 리스트의 쌍이 안맞을 경우 Null값이 할당된다. 루프 횟수는 원소의 갯수가 많은 쪽이 정한다.

	foreach { e1 e2 } { one 1 two  2 three 3 four 4 } {

    puts “$e1 $e2”

}

set data { one 1 two 2 three }

foreach { e1 e2 } $data {

    puts “$e1 $e2”

}




loopVariable과 valueList 쌍이 여러 개 올 수도 있다. 각 리스트의 원소 갯수가 안 맞을 경우 모자라는 쪽이 Null값이 할당된다. 루프 횟수는 원소 갯수가 많은 쪽 리스트에 의해서 정해진다.

	foreach e1 { one two three four } e2 { 1 2 3 4 } {

    puts “$e1 $e2”

}

set list1 [ list one two ]

set list2 { 1 2 3 }

foreach e1 $list1 e2 $list2 {

    puts “$e1 $e2”

}




이를 조합하면 다음과 같은 조금 복잡한 모양도 가능하다.

	foreach name { apple orange } { e1 e2 } { one 1 two 2 } {

    puts “$name : $e1 = $e2”

}


1.2.6. break, continue 명령어
break는 loop를 빠져 나가기 위해 사용된다. 무한 루프를 빠져나가기 위한 프로그램을 다음과 같이 작성할 수 있다.

	while { 1 } {

    ...

    if { $x > 10 } {

        break

    }

    ...

}


다음의 예제에서 볼 수 있듯이 break는 가장 안 쪽 루프만 빠져 나갈 수 있다는 점에 유의하여야 한다.

	for { set ii 0 } { $ii < 3 } { incr ii } {

    puts “ii : $ii”

    for { set jj 0 } { $jj < 5 } { incr jj } {

        if { $jj > 2 } {

            break

        }

        puts “jj : $jj”

    }

}


continue는 continue이하 명령어들을 실행하지 않고 새로운 loop를 시작한다. 

	for { set k -2 } { $k < 2 } { incr k } {

    if { $k == 0 } {

         continue

    }

    set ans [ expr 4 / $k ]

    puts $ans

}


1.2.7. return 명령어
함수를 종료하고 함수를 호출한 바로 다음 줄로 복귀한다. 

	proc mysub { } {

    for { set k 0 } { $k < 10 } { incr k } {

        if { $k > 5 } {

            return

        }

        puts $k

    }

}


Argument가 주어질 경우 그 값을 함수의 리턴 값으로 반환한다.

	proc mysub { } {

    return 3

}


1.2.8. catch 구문
catch body ?errorMessageVariable?

body를 수행하다 error가 발생하면 0이 아닌 값을 리턴한다. 에러 발생 시 리턴되는 값을 보고 에러의 종류를 판별할 수 있다. 에러 발생을 설명하는 메시지를 맨 마지막의 변수에 저장할 수 있다.

직접 함수를 제작할 경우에도 적절한 에러 코드를 리턴하도록 하여 어떤 상황에서 에러가 발생했는지 명확히 해주는 것이 좋다. built-in명령어나 Tcl/Tk인터프리터와 함께 제공된 라이브러리 함수를 이용하는 경우에는 적절히 error를 발생시키도록 프로그램되어 있다. 직접 제작한 함수 내에서 error를 발생시키기 위해서는 “1.2.9

 REF _Ref462291228 \* MERGEFORMAT error 명령어” 절을 참조하기 바란다.

	if [ catch { ... } errorMessage ] {

    puts stderr “Error : $errorMessage”

} else {

    puts “OK! I done”

}


catch구문을 사용하는 가장 대표적인 경우는 file IO함수를 사용하거나 exec명령어를 사용하는 경우이다. 보다 자세한 내용은 “1.8.3

 REF _Ref431802893 \* MERGEFORMAT file명령어”절과 “1.9.1

 REF _Ref462291943 \* MERGEFORMAT exec 명령어”를 참조하기 바란다.

에러가 발생하면 인터프리터가 자동으로 전역 변수 (global variable)인 errorInfo를 셋팅해 준다. 이 변수의 사용법은 다음과 같다. 변수 errorInfo에는 에러가 발생했을 때의 여러 가지 상황에 대한 정보가 담겨 있다.

	if [ catch { ... } errorMessage] {

    global errorInfo

    puts stderr “Error : $errorMessage”

    puts stderr “Below is TCL TRACED Error Informations”

    puts stderr $errorInfo

}


catch구문의 body를 { }가 아닌 [ ]로 묶으면 에러가 발생한다. 그런데 발생하는 에러 메세지가 이러한 사실을 정확히 알려 주지 않고 [ ]로 묶인 명령어에 따라 에러 메세지가 틀리게 나온다. 따라서 이렇게 발생한 버그는 매우 찾기 힘든 버그 중에 하나이다. 

	# 다음과 같이 catch를 사용하면 알 수 없는 에러 발생
if [ catch [ ... ] errorMessage] {

    ...

}


body를 수행하면서 return, break, continue로 body의 수행이 끝난 경우에도 catch문으로 프로그램 흐름이 넘어 온다. 이 때 catch문이 리턴하는 값은 다음과 같다. 리턴 값을 보고 적절히 프로그램의 의도에 맞게 적절히 처리해 주어야 한다. 이러한 기능이 실제 프로그램에서 그렇게 유용한 기능으로 생각되진 않는다. 필자는 아직까지 한번도 아래 기능을 사용해서 프로그램해야 하는 경우가 없었다.

	switch [ catch { ... } message] {

    0 {                    # 정상적인 종료, 즉 에러 없음 }

    1 {                    # 에러가 났음  }

    2 { return $message # return으로 body가 끝난 경우 임}

    3 { break             # break로 body가 끝난 경우 임}

    4 { continue         # continue로 body가 끝난 경우 임}

    default {             # 알 수 없는 경우 임}

}


1.2.9. error 명령어
이전 절에서 catch구문에 대하여 공부하였다. 대부분의 경우에는 catch구문만 어떻게 사용하는지 알고 있어도 충분하다. Tcl 인터프리터가 제공하는 Built-In 라이브러리들은 내부적으로 적절한 곳에서 error명령을 사용하여 error를 발생시키도록 프로그램되어 있기 때문이다. 

catch 구문이 발생한 에러를 처리하는 부분이라면, error 명령어는 에러를 처음으로 발생시키는 명령어이다. 발생된 에러는 catch구문으로 처리되어야 하는데 프로그래머가 이를 처리하지 않을 경우 Tcl인터프리터의 기본 에러 핸들러가 이를 처리하게 된다. Tcl 인터프리터의 기본 에러 핸들러는 발생된 에러에 대한 에러메시지를 출력하고 Script의 실행을 종료시킨다.

error message ?info? ?code?
message는 catch문에서 맨 마지막 변수에 저장될 에러 메시지이다. 함수를 만드는 방법에 대해서는 아직 공부하지 않았지만 다음을 그대로 입력하여 결과를 보면서 error명령어의 동작을 음미해 보기 바란다.

	proc babo { } {

     error “This is Error Message”

}

catch [ babo ] message

set $message


연습문제
1. 구구단을 출력하는 프로그램을 작성 하시오.

2. 다음과 같이 동작하는 Text Mode게임을 작성하시오
*  컴퓨터는 1 부터 60 중 임의의 숫자를 고른다.

*  게임 플레이어는 이 숫자를 맞춘다.

*  플레이어가 숫자를 맞출 때까지 컴퓨터는 플레이어에게 숫자를 입력시킨다.

*  컴퓨터가 고른 숫자보다 플레이어의 숫자가 크면 숫자가 크다는 메시지 출력
*  컴퓨터가 고른 숫자보다 플레이어의 숫자가 작으면 숫자가 작다는 메시지
*  맞췄으면 축하 메시지와 점수를 출력
*  점수는 반복 횟수가 적을 수록 높게
*  임의의 숫자를 구하기 위해서 UNIX명령어인 date와 cut를 사용하시오.

            date | cut -c18-19

*  사용자가 입력한 숫자를 userguess라는 변수로 입력받기 위해 다음과 같이 프로그램하시오
            gets stdin userguess

1.3. 문자열 조작
“오류! 참조 원본을 찾을 수 없습니다.

 REF _Ref431799811 \* MERGEFORMAT 오류! 참조 원본을 찾을 수 없습니다.”에서도 언급하였지만 Tcl과 같은 최신 스크립트 언어들이 프로그래밍 언어에 비하여 각광을 받고 있는 이유 중에 하나는 무엇보다도 문자열을 조작하기 위한 강력한 기능들을 갖추고 있기 때문이다. 일반 프로그래밍 언어로 프로그램을 짜기 위해서는 수백 라인에 걸쳐서 프로그램 해야 할 일들이 Tcl에서는 단 몇 줄의 프로그램만으로도 가능한 일들 중 하나가 바로 문자열 조작이다. 특히 정규 표현식과 같은 기능을 활용한다면 Tcl의 강력한 기능을 더욱 실감할 수 있을 것이다.

1.3.1. Tcl문자열에 대한 정의
Tcl에서 문자열은 알파벳이나 밑줄문자로 시작하는 글자들의 집합이다. 알파벳이나 밑줄문자로만 시작하면 그 다음에 올 수 있는 문자들에는 제한이 없다. 단, Tcl 인터프리터가 특별한 의미를 부여한 기호를 문자열에 포함시키기 위해서는 백슬래쉬( \ )를 사용해야 한다. 여기에 관한 자세한 내용은 “1.1.7

 REF _Ref430745223 \* MERGEFORMAT 백슬래쉬 문자열( Backslash Sequence)”을 참조하기 바란다.

문자열에 공백 문자를 포함시키기 위해서는 따옴표를 사용하면 된다. 공백문자가 없는 문자열의 경우는 따옴표로 감싸줄 필요가 없다.

1.3.2. string 명령어
string명령어는 문자열에 관해서 할 수 있는 많은 연산들을 한데 모아 놓은 명령어 이다. 이 명령어의 기능을 요약하면 다음과 같다.

	문법
	기능

	string compare str1 str2
	str1과 str2를 사전 순으로 비교하여, 같으면 0, str1이 str2 보다 앞서면 -1, 뒤서면 1을 리턴

	string first str1 str2 
	str1이 str2에서 첫번째로 나타나는 인덱스를 리턴

	string index str i
	str의 i번째 문자를 리턴


	string last str1 str2
	str1이 str2에서 마지막으로 나타나는 인덱스를 리턴

	string length str
	str의 길이를 리턴

	string match pattern str
	glob style패턴
이 str에 매칭되면 1을 아니면 0을 리턴

	string range str i j
	str의 i번째에서 j번째까지의 문자들을 리턴

	string tolower str
	str내의 대문자를 모두 소문자로 바꿈

	string toupper str
	str내의 소문자를 모두 대문자로 바꿈

	string trim str ?chars?
	str의 시작 부분과 끝 부분에 존재하는 chars를 없애버린다. chars는 따로 지정하지 않으면 공백문자들이다.

	string trimleft str ?chars?
	str의 시작 부분에 존재하는 chars를 없애 버린다.

	string trimright str ?chars?
	str의 끝 부분에 존재하는 chars를 없애 버린다.

	string wordend str ix
	str내에서 ix번째 글자가 포함 된 단어가 끝나는 지점의 인덱스를 리턴. 단어 사이는 공백 문자로 구분한다.

	string wordstart str ix
	str내에서 ix번째 글자가 포함 된 단어가 시작하는 지점의 인덱스를 리턴


표 4 string명령어 기능 요약
string명령어의 실제 사용 예를 보이면 다음과 같다.

	% set str ":::::: won nara is genius :::: nara? really? ::::"

:::::: won nara is genius :::: nara? really? ::::

% string first "nara" $str

11

% string last "nara" $str

31

% string index $str 1

:

% string length $str

49

% string match ":*" $str

1

% string trim $str :

 won nara is genius :::: nara? really?

% string trimleft $str :

 won nara is genius :::: nara? really? ::::

% string trimright $str :

:::::: won nara is genius :::: nara? really?

% string toupper $str

:::::: WON NARA IS GENIUS :::: NARA? REALLY? ::::

% string range $str 11 14

nara

% string wordend $str 12

15

% string wordstart $str 12

11

%


1.3.2.1. 문자열 매칭
string match명령어에서 사용되는 패턴은 glob-style패턴이다. 이는 UNIX 쉘에서 사용하는 wild card문자들을 사용하여 기술한 패턴을 말한다. UNIX 사용자라면 이 패턴에 익숙해 있을 것이다. glob-style패턴 기술에 사용되는 wild card 문자들은 과 같다.

	Wild card
	의미

	*
	아무 글자에나 매칭이 되고 0개 이상의 글자에 매칭된다.

	?
	아무 글자나 한 글자에만 매칭 된다.

	[ ]
	[ 와 ] 사이에 열거된 문자에만 매칭 되고 한 글자만 적용된다.



다음의 사용 예를 통하여 glob-style패턴 사용법을 정확히 이해하도록 하자. 패턴은 “{ }”로 싸줘야 제대로 동작한다.

	% string match {a*} about

1

% string match {a?} about

0

% string match {a????} about

1

% string match {[aA]} a

1

% string match {[aA]} A

1

% string match {[aA]} aA

0

% string match {[aA][aA]} aA

1

% string match {a\?} a?

1


1.3.2.2. 문자열 비교
Tcl에서 문자열을 비교하는 가장 확실한 방법은 다음과 같이 string compare명령어를 이용하는 방법이다.

	if { [ string compare $var MODEL ] == 0  } {

     ...

}

...

if { [ string compare $var “Test String” ] == 0 } {

    ...

}

...

if { [ string compare $var1 $var2 ] == 0 } {

    ...

}


이 방법 말고도 “ == ” 연산자를 사용하여 다음과 같이 두 문자열이 서로 같은 지 비교할 수 있다.

	if { “$var” == “MODEL” } {

    #변수와 문자열을 비교할 때는
    #변수와 문자열을 반드시 따옴표로 묶어야 Syntax Error가 발생하지 않음
    ...

}

...

set a “HaHa HoHo”

set b “HaHa HoHo”

if { $a == $b } {

    #변수끼리 비교할 때는 따옴표로 묶을 필요 없음
    ....

}




1.3.3. format 명령어
format명령어는 어떤 형식으로 변수를 문자열로 바꿔줄 것인가를 명시하는 명령어이다. C언어에서 printf()함수를 사용해본 경험이 있다면 이 명령어를 쉽게 이해할 수 있을 것이다. 

format spec value1 ?vlaue2 ...?

C언어의 다음 문장을 똑같은 결과가 나오도록 Tcl로 프로그램한 예는 다음과 같다.

	/* C언어 문장 */

printf( “Sum is %5.2f\n”, sum );



	# Tcl문장
puts [ format “Sum is %5.2f” $sum ]




spec은 문자열이며 변수의 값을 찍어 줄 자리에 %기호를 이용하여 변수를 문자열로 변환할 규칙을 명시한다. spec문자열 안에서는 “%”가 특별한 의미를 갖고 있기 때문에 “%” 기호 자체를 문자열에 포함시키고 싶을 경우에는 “%%”를 사용해야 한다. puts 명령어를 이용하여 “%”를 찍을 때는 하나만 사용해도 되지만 format 명령어의 spec 문자열 안에서 %를 찍기 위해서는 “%%”라고 써야 한다. %문은 다음과 같은 필드들로 구성된다.

%?flags??preceeding??width??precision?type

width필드는 변수가 문자열로 변환될 공간의 최소 크기를 표시한다. 변환된 문자열이 이 크기 보다 작아도 width 만큼 공간을 차지하도록 한다. 만일 변환된 문자열의 길이가 width보다 크면 필요한 만큼 공간을 늘인다. width는 숫자로 명시하는 것이 일반적이지만 그 크기를 프로그램 작성시에는 알 수 없고 프로그램 동작 중에 명확해 지는 경우도 있다. 이를 위하여 width필드에 “*”를 쓰고 width값을 뒤에 변수로 줄 수도 있다.

precision필드는 실수를 문자열로 변환하고자 할 때 필요한 필드이다. 구두점 “ . ” 다음에 소수점 이하 몇 째 자리까지 변환할 것인가를 명시한다. 명시된 것보다 작은 부분은 반올림 된다.

type필드는 변수에 저장된 값을 무엇으로 해석해야 할 것인가를 알려주는 필드이다. 이 필드는 “%”기호 다음에 반드시 존재해야 한다. 여기에 가능 값들과 의미를 표 5에 정리하였다.

	Type
	의미

	d
	정수

	u
	양의 정수

	I
	정수, 변수에 기억된 값이 8진수(0), 16진수(0x)형식으로 저장된 문자열일 때 사용한다.

	O
	양의 8진수

	x
	양의 16진수, 알파벳이 필요할 경우 소문자로 사용

	X
	양의 16진수, 알파벳이 필요할 경우 대문자 사용

	c
	변수에 저장된 숫자를 ASCII코드로 생각하고 그에 해당하는 글자로 변환 시킴

	s
	문자열

	f
	실수, 구두점으로 구분된 형식 (예: 3.14)

	e
	실수, 과학적 표기법, e를 사용 (예: 1.23e14)

	E
	실수, 과학적 표기법, E를 사용 (예: 1.23E14)

	g
	실수를 %f나 %e 중 짧은 쪽으로 변환

	G
	실수를 %f나 %E 중 짧은 쪽으로 변환


표 5 format명령어 %구문 type필드
flags필드는 변환된 문자열의 모양을 다듬기 위해 사용되는 필드이다. 이 필드에 사용할 수 있는 플래그들을 표 6에 정리하였다.

	Flag
	의미

	-
	변환된 문자열의 길이가 width보다 작을 경우 왼쪽으로 정렬 한다.

	+
	숫자의 경우 양, 음수 부호를 반드시 출력한다.

	#
	이 플래그는 type필드에 따라서 다른 의미를 갖는다. 아래의 type들 이외에는 아무런 의미가 없다.

·  %o : 맨 앞에 항상 0을 붙인다.

·  %x : 맨 앞에 항상 0x를 붙인다.

·  %g : 소수점 이하에서 뒤에 붙는 0들을 없애지 않는다.


표 6  format명령어 %구문 플래그
preceeding 필드에 올 수 있는 것은 0과 스페이스 문자이다. width에 정의된 공간 보다 숫자가 차지하는 문자열의 길이가 짧을 경우 왼쪽에 이 문자들로 적당히 채워 width의 길이가 되도록 한다. 왼쪽으로 정렬시켰을 때는 아무 의미도 없다.

다음에 format 명령어의 사용 예를 음미하면서 format 명령어의 기능을 정확히 이해하도록 하자.

	% format "%010d" 33

0000000033

% format "%03.3f" 1.23456

1.235

% format "%010.3f" 1.23456

000001.235

% format "%o" 8

10

% format "%#o" 8

010

% format "%x" 10

a

% format "%X" 10

A

% format "%#X" 10

0xA

% format "%e" 1.23456

1.234560e+00

% format "%E" 1.23456

1.234560E+00

% format "%g" 1.234

1.234

% format "%g" 0.00000000000000012343

1.2343e-16

% format "%g" 1.200000

1.2

% format "%#g" 1.20000

1.20000

% format "%*s" 10 aaa

       aaa

% format "%0*d" 5 3

00003

% puts [format "Result : %s = %d %%" value 30]

Result : value = 30 %


1.3.4. scan 명령어
scan명령어는 주어진 문자열을 명시한 spec에 따라 해석하는 기능을 한다. 리턴 값은 해석을 해서 변환한 갯수이다. 문자열로 표현된 숫자를 계산에 사용될 수 있는 숫자로 변환하거나 글자를 ASCII 코드로 변환하는 경우 등에 유용하게 사용될 수 있다. format명령어와 반대되는 기능을 한다고 볼 수 있다. C언어에서 scanf() 함수와 비슷한 기능을 한다.

scan str spec var1 ?var2 ...?

scan에서 사용하는 spec은 format에서 사용하는 spec과 거의 같다. 단, %u를 사용할 수 없으며 “[ ]”를 사용하여 가능한 문자 집합을 명시할 수 있다.
 “[ ]”를 사용할 경우 따옴표 안에서 사용하면 따옴표는 치환을 허용하기 때문에 “[ ]”를 명령어 netsting으로 해석하려 한다. 이를 방지하기 위해 치환을 허용하지 않는 중괄호를 사용해서 spec을 명시해야 한다.

	% scan “0xA 010” “%x %o” hex oct

error

% scan “A 10” “%x %o” hex oct

2

% set hex

10

% set oct

8

% scan “abcABCdef” “%[a-z]” low

invalid command name "a-z"

% scan “abcABCdef” {%[a-z]} low

1

% set low

abc


1.3.5. 문자열 덧붙이기와 분할, 병합
문자열을 분할하고 병합하는 함수는 리스트와 밀접한 관계를 갖고 있다. 리스트에 관한 자세한 내용은 “1.4

 REF _Ref430745568 \* MERGEFORMAT 리스트(List)”절을 참조하기 바란다.

1.3.5.1. append 명령어
append 명령어는 첫번째 Argument로 주어진 문자열 변수에 저장된 문자열과 나머지 Argument들을 연결하여 하나의 문자열로 만들어 준다.

	% set fruit “app”

app

% append fruit le

apple

% set fruit

apple

%


1.3.5.2. split 명령어
필드 분리자(field separator)를 명시해 주면 그 필드 구분자로 분리된 필드들을 추출하여 리턴한다.

	% set test_string host:address:phone:name

host:address:phone:name

% split $test_string :

host address phone name


1.3.5.3. join 명령어
주어진 리스트의 각 원소들 사이에 주어진 필드 분리자(field separator)를 넣어 하나의 문자열로 병합한다.

	% set myData { host address phone name }

host address phone name

% join $myData :

host:address:phone:name

% 


1.4. 리스트(List)

Tcl에서 리스트는 서로 관련 있는 것을 그룹화 시켜 놓고 한꺼번에 다루고자 할 때 매우 유용한 데이터 구조이다. 

foreach 문을 이용하여 리스트의 모든 원소들에 대하여 목적에 맞는 특별한 일을 할 수 있다. 

eval 명령어를 이용하면 리스트에 기억된 내용들을 Tcl 명령어로 생각하고 재해석하여 실행하도록 할 수 있다. eval 명령어는 차후에 수행될 명령어를 미리 만들어 기억시켜 두었다가 적당한 시기에 수행시키고자 할 때 유용하다.

1.4.1. 리스트의 생성
리스트는 중괄호를 이용하여 생성할 수도 있으며 list명령어를 이용하여 생성할 수도 있다. 중괄호를 이용하는 것 보다 list 명령어를 이용하는 편이 수행 속도가 빠르다. 명령어의 Argument로 리스트를 받아들이는 명령어에서 중괄호를 사용해서 리스트를 만들어 넘길 때 에러가 나는 경우 list명령어로 리스트를 만들어 넘겨주면 에러를 없앨 수 있다.

	% set a { 1 2 3 4 }

1 2 3 4

% set b [ list a b c d ]

a b c d

demo

haha


공백 문자를 포함하는 리스트 원소를 정의하기 위해서는 중괄호나 따옴표를 이용한다. 

	% set a { “1 2” { 3 4 } 5}

“1 2” {3 4} 5

% set b [ list a {b c} “d e” ]

a {b c} {d e}

% foreach x $a {

     puts $x

}

1 2 

3 4

5

% foreach x $b {

    puts $x

}

a

b c

d e


그러나 Tcl은 공백 문자를 리스트 원소를 구별하는데 이용하기 때문에 공백 문자를 포함하는 리스트 원소는 또 하나의 리스트를 구성한다.

	% set b [list {a b c} d]

{a b c} d

% foreach x $b {

     foreach c $x {

        puts $c

     }

}

a

b

c

d


1.4.2. 리스트 추가 및 연결
리스트 관련 명령어들을 공부할 때 주의해야 할 점은 각 명령어가 원본 변수의 값을 변화 시키는가 아닌가를 눈여겨 볼 필요가 있다. 보통 변수의 이름을 Argument로 받는 경우는 원본 변수의 값을 변화 시키고, 변수의 값을 Argument로 받는 경우엔 원본 변수의 값을 변화시키지 않는다. 이러한 규칙은 모든 Tcl명령어에 해당된다. 간단히 변수 이름 앞에 $가 붙으면 원본이 변하지 않고, $가 붙지 않으면 원본이 변한다고 생각하면 쉽다. 원본 변수의 값이 변하지 않는 경우에는 명령어의 리턴값을 적당한 변수에 저장하여 사용하여야 한다.

1.4.2.1. lappend 명령어
존재하지 않는 변수가 주어졌을 때는 새로 리스트를 생성하고 이미 존재하는 변수가 주어졌을 때는 그 변수 리스트의 끝에 리스트를 덧붙인다. 원본 변수의 값이 바뀐다.

	% lappend myList 1 2

1 2

% lappend myList 3 {4 5}

1 2 3 {4 5}

% lappend myList 6 7 8 9

1 2 3 {4 5} 6 7 8 9

% set myList

1 2 3 {4 5} 6 7 8 9


1.4.2.2. concat 명령어
주어진 Argument들을 연결하여 하나의 리스트로 만든다. 원본 변수의 값이 바뀌지 않는다.

	% set a [list 1 2]

1 2

% set b [list 3 4]

3 4

% set c 5

5

% concat $a $b $c 6 7

1 2 3 4 5 6 7


1.4.3. 리스트 원소 얻기
본 절에서 설명하고 있는 lindex와 lrange명령어를 사용할 때 리스트에 저장된 데이터가 매우 많은 경우 약간의 주의를 요한다. Tcl 8.0이전 버전에서는 리스트의 크기가 매우 큰 경우, 이 들 명령어의 수행 속도가 현저히 떨어지기 때문이다. Tcl 8.0에 와서 개선 되기는 하였으나 리스트에 저장할 데이터가 매우 많을 경우에 lindex와 lrange명령어를 사용해야 한다면 배열을 사용할 것을 권장한다.
 배열에 대한 자세한 내용은 “1.5

 REF _Ref430745645 \* MERGEFORMAT 배열”을 참조하기 바란다.

1.4.3.1. llength 명령어
주어진 리스트의 원소 개수를 리턴한다.

	% llength { a b c d }

4

% llength { {a b} “c d” e}

3

% llength {}

0

% set kk [ list a b c ]

a b c

% llength $kk

3


1.4.3.2. lindex 명령어
주어진 리스트 중에서 인덱스에 해당하는 원소를 리턴한다. 인덱스는 0부터 시작한다.

	% lindex { a b c d } 1

b

% set k [list a b c d ]

a b c d

% lindex $k 0

a

% 


1.4.3.3. lrange 명령어
주어진 리스트 중에서 시작 인덱스와 끝 인덱스 사이의 원소를 리턴한다. 리스트의 끝까지를 나타내는 특별한 인덱스 end가 있다
	% set k [list a b c d e f g]

a b c d e f g

% lrange $k 1 4

b c d e

% lrange $k 4 end

e f g

%


1.4.4. 리스트 원소 삽입 및 치환
리스트 원소 삽입 및 치환 명령어들은 대부분 원본 변수 값을 바꾸지 않는다는 점에 유의하기 바란다. 삽입되거나 치환된 결과를 이용하기 위해서는 명령어의 리턴값을 따로 저장해야 한다.

1.4.4.1. linsert 명령어
.

리스트에 Argument들을 주어진 인덱스 위치에 끼워 넣는다.

	% linsert { 0 1 2 3} 2 a b c d

0 1 a b c d 2 3

% set k [list a b c d]

a b c d

% linsert $k 0 5 6

5 6 a b c d

%


1.4.4.2. lreplace 명령어
리스트에서 시작 인덱스부터 끝 인덱스까지 주어진 Argument들로 바꾼다.

	% lreplace { 0 1 2 3 4 } 1 3 a b c d e

0 a b c d e 4

% lreplace { 0 1 2 3 4 } 2 end d

0 1 d


1.4.5. 리스트 검색과 정렬
1.4.5.1. lsearch 명령어
리스트의 원소 중에서 패턴에 맞는 첫번째 원소를 찾아 그 원소를 리턴한다. 리스트 내에 패턴에 맞는 원소가 없을 경우 -1을 리턴한다.  이 명령에 가능한 옵션들은 다음과 같다.

*  -glob : glob스타일의 패턴 매칭
을 한다. 옵션이 없는 경우 이 옵션으로 지정되어 있다고 생각한다.

*  -exact : 정확히 모든 글자가 일치하는 경우
*  -regexp : 정규 표현식
을 이용한 패턴 매칭을 한다.
	% lsearch { apple air snow } a*

0 

% lsearch -exact { apple air snow } apple

0

% set kk { test haha hello }

test haha hello

% lsearch $kk apple

-1


1.4.5.2. lsort 명령어
리스트의 원소들을 정렬하여 리턴한다. 단, 리스트 자체는 변하지 않는다. 가능한 옵션들은 다음과 같다.

* -ascii : 리스트 원소들을 ASCII값에 따라 정렬한다.

* -dictionary : 리스트 원소들을 사전 순으로 정렬한다.

* -integer : 리스트 원소들이 모두 정수임을 명시한다. 

          정수가 아닌 경우 에러가 발생한다.

* -real : 리스트 원소들이 모두 실수임을 명시한다. 

        실수가 아닌 경우 에러가 발생한다.

* -increasing : 오름순으로 리스트를 정렬한다. 디폴트 값이다.

* -decreasing : 내림순으로 리스트를 정렬한다.

	% lsort -ascii { Z m b B }

B Z b m

% lsort -dictionary { Z m b B }

B b m Z

% set k [list 7 3 5 1 ]

7 3 5 1

% lsort $k

1 3 5 7

% set k

7 3 5 1


연습문제
다음과 같은 데이타 파일을 읽어들여 그래프를 그리고자 한다. 그래프를 그리는 것은 Tk를 공부한 다음으로 잠시 미루고, 그래프를 그리기 위한 전 단계로 다음과 같은 데이타 파일이 주어졌을 때 이를 해석하여 다음과 같은 출력을 보여 주는 프로그램을 작성한다.

프로그램을 작성할 때는 Tk공부 후 그래프를 그릴 수 있도록 데이타 구조를 설계하고 구현하여야 한다. 단, 현재까지 File입출력을 다루지 않았으므로 stdin에서 데이타를 입력 받는 것으로 프로그램을 작성한 후 UNIX와 Windows에서 제공하는 File Redirection을 이용해서 데이타 파일을 입력 받도록 한다. 예를 들어 작성한 프로그램 이름이 myprog2.tcl이고 데이타 파일 이름이 mydata.dat이라면 다음과 같이 실행한다.

$ myprog2.tcl < mydata.dat

파일의 끝인지 테스트 하는 Tcl 명령어에는 eof가 있다. stdin으로 입력되는 파일의 끝인지를 테스트하기 위해서는 다음과 같이 하면 된다.

while { [ eof stdin ] == 0 } {

데이타 파일
	#

# This is sample data file for Tcl/Tk programming project

#

TITLE           : Imaginary data for test

COMMENT         : 1999.2.9

DATA_NUM        :       4

LOT_NUM         :       5

FIELD_NAME      :    LOT1 LOT2 LOT3 LOTP LOTK

MPU                    100  200   400 500   20

DSP                    10   50     80 90    100

PKG                    300  400   320 430   650

TGET                   400  210   330 410   200


화면 출력
	title       : Imaginary data for test

comment     : 1999.2.9

data number : 4

lot number  : 5

field names  : LOT1 LOT2 LOT3 LOTP LOTK

DATA : MPU - 100 200 400 500 20

DATA : DSP - 10 50 80 90 100

DATA : PKG - 300 400 320 430 650

DATA : TGET - 400 210 330 410 200


1.5. 배열
배열은 일련의 데이터를 인덱스를 사용하여 저장하고 참조할 수 있도록 해준다. C언어와 같은 프로그래밍 언어에서와는 달리 인덱스로 문자열을 사용한다. 따라서 Tcl배열의 문자열 인덱스를 잘 활용하면 이진 트리나 변환 테이블 같은 복잡한 데이터 구조를 쉽게 만들 수 있다.

1.5.1. 배열의 생성과 참조
배열은 배열의 한 원소에 값을 할당할 때 배열이 존재하지 않으면 생성된다. 값을 할당하지 않은 배열 원소를 참조할 경우에는 에러가 발생한다. 다음 예제를 통하여 배열을 생성하고 참조하는 방법을 이해하도록 한다.

	% set myname(last) won

won

% set lastname $myname(last)

won

% set firstname $myname(first)

can't read "myname(first)": no such element in array

% set myname(first) nara

nara

% set firstname $myname(first)

nara

% set fruit(0) apple

apple

% set fruitname $fruit(0)

apple

% set index last

last

% puts $myname($index)

won


배열을 사용할 때 잦은 실수는 다음과 같은 경우이다.

	% set myname(myindex) won

won

% set index myindex

myindex

% puts $myname( $index )

error


여기서 명심해야 할 것은 배열의 인덱스는 문자열이기 때문에 공백 문자도 인덱스를 나타내는 문자열로 해석된다는 점이다. 위의 예제에서 괄호 사이에 공백 문자가 존재하기 때문에 Tcl 인터프리터는 인덱스 문자열을 공백을 포함한 문자열인 “ myindex ”을 가지고 배열의 원소를 찾게 되므로 에러가 발생하는 것이다. 이러한 에러를 방지하기 위해서는 다음과 같이 수정해야 한다.

	% set myname(myindex) won

won

% set index myindex

myindex

% puts $myname($index)

error


따라서 배열의 인덱스를 표기할 때 괄호와 인덱스 사이에 공백이 존재하지 않도록 해야 한다는 점을 명심해야 한다. 

1.5.2. array명령어
array명령어는 배열에 대한 여러 가지 정보를 리턴한다. array명령어의 기능을 요약하면 표 7과 같다.

	문법
	기능

	array exists arr
	배열 arr이 존재하면 1을 리턴 한다.

	array get ?pattern?
	배열 내의 모든 원소들에 대해 인덱스, 값의 순서로 리스트를 만들어 리턴한다. 패턴이 주어지면 패턴에 일치하는 인덱스에 대해서만 리스트를 만든다. 

예 : { 0 apple 1 orange } 

	array names arr ?pattern?
	배열 내의 모든 인덱스의 리스트를 리턴한다. 패턴이 주어진 경우 패턴에 일치하는 인덱스들만 리스트로 만든다.

	array set arr list
	array get에서와 같은 형식으로 된 리스트를 배열로 만든다.

	array size arr
	배열에 정의된 인덱스의 갯수를 리턴한다.

	array startsearch arr
	배열 arr에 대한 검색 작업을 초기화 하고 이 작업을 identify하는 ID를 리턴한다.

	array nextelement arr id
	배열 arr에 대하여 행해지고 있는 검색작업 id에서 다음 인덱스를 리턴한다. 더 이상의 인덱스가 없으면 빈 문자열을 리턴한다.

	array anymore arr id
	배열 arr에 대하여 행해지고 있는 검색작업 id에서 남아 있는 인덱스가 더 있으면 1을 리턴한다.

	array donesearch arr id
	배열 arr에 대한 검색 작업 id를 끝마친다.


표 8 array명령어의 기능 요약
array의 모든 인덱스에 대하여 어떤 일을 하는 프로그램을 작성해 보면 다음과 같다. 

	foreach index [array names myArray] {

     puts $myArray($index)

}


똑같은 일을 하는 프로그램을 아래와 같이 작성할 수도 있다.

	set id [array startsearch myArray]

while {[array anymore myArray $id]} {

     set index [array nextelement myArray $id]

     puts $myArray($index)

}

array donesearch myArray $id


array names 명령어에는 원하는 패턴을 명시하여 패턴과 매칭되는 인덱스만을 골라 특별한 일을 처리할 수도 있음을 눈여겨 보자. 이를 이용하면 보다 프로그램을 간단히 구현할 수 있다.

	foreach index [array names myArray “.signal.” ] {

     puts $myArray($index)

}


1.5.3. 배열을 이용한 복잡한 데이터 구조 구현
배열을 이용해서 구현할 수 있는 데이터 구조는 무궁무진하다. 몇 가지 예를 통하여 배열로 복잡한 데이터 구조를 만들기 위한 힌트를 얻어 보자.

1.5.3.1. 주소록
	proc AddAddress { name address } {

    global array TABLE

    set TABLE($name) $address

}

proc GetAddress { name } {

    global array TABLE

    return $TABLE($name)

}




1.5.3.2. 이진 트리
다음 프로그램은 그림 1와 같은 이진 트리를 구성한 후 Top-Down방식으로 왼쪽부터 오른쪽으로 트리를 여행하면서 각 노드의 값을 찍어 주는 프로그램이다.




그림 1 트리 구조
	#!/usr/local/bin/tclsh

proc MakeNewNode { name value } {

     global tree

     set tree($name.value) $value

     set tree($name.left) empty

     set tree($name.right) empty

}

proc AddNodeLeft { parent child } {

    global tree

    set tree($parent.left) $child

}

proc AddNodeRight { parent child } {

    global tree

    set tree($parent.right) $child

}

proc Tour { startnode } {

    global tree

    puts $tree($startnode.value)

    if { “$tree($startnode.left)” !=  “empty” } {

        Tour $tree($startnode.left)

    }

    if { “$tree($startnode.right)” != “empty” } {

        Tour $tree($startnode.right)

    }

}

MakeNewNode node1 apple

MakeNewNode node2 lion

MakeNewNode node3 rose

MakeNewNode node4 orange

MakeNewNode node5 tree

AddNodeLeft  node1 node2

AddNodeRight node1 node3

AddNodeLeft  node2 node4

AddNodeRight node3 node5

  Tour node1


연습문제
1. 배열을 이용하여 Stack을 구현하시오.

2. 배열을 이용하여 Queue를 구현하시오.

3. 배열을 이용하여 다차원 배열을 구현하시오.

1.6. 정규 표현식
정규 표현식 ( Regular Expression)은 원하는 문자열을 찾아내어 적절히 조작하기 위해 사용되는 강력한 문법이다. 문자열의 Pattern Matching을 위해 사용되는 문법이라고 이해하면 된다.

1.6.1. 정규 표현식 Premitives

정규 표현식에 사용되는 연산자들을 표 9에 정리하였다. 단, **표시가 된 기능은 Tcl/Tk 8.1 이상의 버전에서만 동작한다.

	종류
	기호
	의미

	메타 (Meta)

 문자
	.
	어떤 글자든지 한 글자와 매칭된다.

	
	^
	문자열의 시작 부분임을 나타냄

	
	$
	문자열의 끝 부분임을 나타냄

	반복
 연산자
	*
	바로 앞 패턴이 0번 이상 반복 됨을 표시

	
	+
	바로 앞 패턴이 1번 이상 반복 됨을 표시

	
	?
	바로 앞 패턴이 나타나지 않거나 1번 나타날 수 있음을 표시

	
	{m}
	바로 앞 패턴이 m번 반복됨을 표시 **

	
	{m,}
	바로 앞 패턴이 m번 이상 반복됨을 표시 **

	
	{m,n}
	바로 앞 패턴이 m번 이상 n번 이하 반복됨을 표시 **

	그룹
	( )
	패턴의 한 부분을 그룹화한다. 패턴에 매칭되는 부분을 기억한다.

	열거 

및 

선택
	[ ]
	[]안에 열거된 글자 중 하나와 같으면 매칭 된 것으로 본다.

	
	[^ ]
	[^와 ]사이에 열거된 글자가 아닌 다른 글자이면 매칭된 것으로 본다.

	
	|
	“ | ” 기호 왼쪽에 명시된 패턴이나 오른 쪽에 명시된 패턴 중 하나에 매칭되면 매칭 된 것으로 본다.

	복귀
	\
	연산자로 사용되는 문자를 패턴에 사용하기 위해 바로 앞에 붙인다.

	Backslash Escape
	
	공백 문자, 숫자 등을 나타내는 패턴과 같이 자주 사용되는 패턴들을 간단히 표현할 수 있도록 여러 가지 backslash escape를 제공한다. **


표 9 정규표현식 연산자
구두점 “.”은 어떤 글자이든 한 글자에 매칭된다. alphanumeric 글자 뿐 아니라 탭, 스페이스, 특수 문자 등 모든 한 문자에 매칭된다. 4글자 이상으로 이루어진 문자열을 찾는 패턴은 다음과 같다. 밑줄 친 부분이 실제로 패턴에 매치되는 부분이다. 문자열안에 패턴과 매치되는 부분이 있기만 하면 그 문자열은 패턴과 매치된 것으로 본다.

	패턴
	....

	매치되는 문자열
	aaaa
1234567

a13b456ba is woow

cat is yellow

	매치되지 않는 문자열 
	bcd

12

a


소문자 알파벳과 숫자로만 이루어진 2글자가 들어 있는 패턴을 찾기 위해 필요한 정규 표현식은 다음과 같다.

	패턴
	[a-z,0-9][a-z,0-9]

	매치되는 문자열
	b9
1a567

BBAa3456ba is woow

NUMBER is 12

THIS IS NUM12

	매치되지 않는 문자열 
	AB1

aABCD

9 IS NOT ALPHABET


“[ ]”는 그 사이에 나열된 모든 글자 중 하나와 일치하면 일치하는 것으로 본다. “a-z”는 “a”에서 “z”까지 모든 문자를 나타낸다. 마찬 가지로 “0-9”도 “0”에서 “9”까지 모든 문자를 나타낸다. “[ ]”사이에 정규 표현식에 사용되는 연산자나 기타 의미 있는 기호들은 모두 단순히 패턴으로만 사용된다. 단, “ [ ” 바로 다음에 “ ^ ”가 오면 열거된 문자가 아닌 문자와 매칭시킨다는 특별한 의미를 갖는다.

	패턴
	AA[^a-z,0-9]

	매치되는 문자열
	AAA
AAZa67

BBAAC456ba is woow

AA9AAD AAA 12

THIS IS AAaAAF

	매치되지 않는 문자열 
	AAa

AA9


“+”는 바로 앞 패턴이 한 번 이상 반복되는 것을 의미한다. 따라서 대문자 알파벳으로만 이루어진 문자열을 포함하는 패턴을 찾기 위한 정규 표현식은 다음과 같이 쓸 수 있다.

	패턴
	[A-Z]+

	매치되는 문자열
	ABCDEFGHIJKLMN
abcdAAF AAA 12

1234THISAFabcd is hahaha

	매치되지 않는 문자열 
	abcd

123

this is not matched


“*”연산자는 0번 이상 패턴이 반복됨을 나타낸다. “+”연산자는 그 패턴이 반드시 나타나야 매칭이 이루어 지지만 “*”연산자는 패턴이 나타나지 않아도 매칭이 이루어진다. 이를 이용하면 앞에 “A”가 반드시 하나 있고 그 뒤에는 대문자 알파벳으로만 이루어진 단어를 포함하는 패턴을 찾기 위한 정규 표현식을 다음과 같이 쓸 수 있다.

	패턴
	A[A-Z]*

	매치되는 문자열
	A

ABCDEFGHIJKLMN
abcdAAF AAA 12

1234AHISAFabcd is hahaha

	매치되지 않는 문자열 
	abcd

123

BCDEF

this is not MATCHED


{}는 반복의 횟수를 지정한다. {m}이라고 쓰면 바로 앞 패턴이 정확히 m번 반복됨을 나타낸다.

	패턴
	a{3}

	매치되는 문자열
	aaa
aaaaaaaaa

123aa345aaaaa

	매치되지 않는 문자열 
	a

aa


{m,}이라고 쓰면 바로 앞 패턴이 m번 이상 반복 됨을 나타낸다.

	패턴
	a{3,}

	매치되는 문자열
	aaa

aaaaa
aaaaaaaaaaaaaaa
123aa345aaaaa

	매치되지 않는 문자열 
	a

aa


{m,n}이라고 쓰면 바로 앞 패턴이 m번 이상 n번 이하 반복 됨을 나타낸다.

	패턴
	a{3,5}

	매치되는 문자열
	aaa

aaaa

aaaaa
aaaaaaaaa

123aa345aaaaaaaa

	매치되지 않는 문자열 
	a

aa


“^”기호는 라인의 처음을 의미한다. “ [^ ”처럼 사용될 때와는 전혀 다른 의미임에 주의하자.

	패턴
	^#.*

	매치되는 문자열
	#
# This is all matched
#AC comment

	매치되지 않는 문자열 
	   # AAa

aaa # this


“$”기호는 라인의 끝을 의미한다.

	패턴
	end$

	매치되는 문자열
	end
aaa # thisend

	매치되지 않는 문자열 
	end is

hahaha end.


“?”는 바로 앞 패턴이 없거나 한번만 나타나는 경우를 나타낸다. 이것을 이용하여 양의 정수를 찾는 정규 표현식을 써보면 다음과 같다. 여기서 “\”는 “+”를 정규 표현식의 연산자로 사용하는 것이 아니라는 것을 명시하기 위해 사용되었다.

	패턴
	\+?[0-9]+

	매치되는 문자열
	2

23456

+123456897

-87.99

-1234 123 is

----+123454abcd is hahaha

+++++123454abcd

abcd123454abcd is hahaha

	매치되지 않는 문자열 
	abcd

BCD

abCD


“( )”는 패턴을 그룹으로 만들어 패턴 전체에 대한 연산이 가능하도록 한다. 그리고 “( )”안의 정규 표현식에 매칭된 패턴을 기억 시켜 두었다가 프로그램의 다른 부분에서 이용할 수 있도록 해 준다. 이것에 관해서는 다음 절에서 자세히 살펴 보기로 한다. “( )”를 이용하여 실수를 찾아 내는 정규 표현식을 써보면 다음과 같다. 구두점 “ . ” 앞의 “\”도 구두점이 정규 표현식의 메타 문자로 사용된 것이 아니라는 것을 나타내고 있다. 조금 복잡하지만 그 의미를 잘 따져 보면 실수의 여러 가지 표현 방식을 모두 나타내고 있음을 알 수 있을 것이다.

	패턴
	[+-]?[0-9]*\.[0-9]+([eE][+-]?[0-9]+)?

	매치되는 문자열
	23456.12

+123456897.34

-456.124

+.42

-.64

.23

45.24e98

32.34e-134

.23e+54

436.34E-99

-22.54E+77

+3243+234.34e-19+1239

abcde154.34-1239AAA

Mean is -456.124 and Sum is 345.34

	매치되지 않는 문자열 
	345

-21


“ | ”는 “OR”의 의미로 사용된다. 즉, 왼쪽의 정규식 혹은 오른쪽의 정규식과 일치하면 일치하는 것으로 간주한다는 의미이다. “ | ”를 사용하여 음수와 양수를 모두 찾을 수 있는 정규 표현식을 써보면 다음과 같다.

	패턴
	(\+|-)?[0-9]+

	매치되는 문자열
	234

+12345

-45

aaa+234+1239

Sum is +54 and

	매치되지 않는 문자열 
	AAA

abCD


정규 표현식을 적용하는데는 2가지 규칙이 있다.

 First Rule

 Greedy Rule

First Rule이란 한 문자열 내에서 패턴과 매칭되는 부분이 여러 개 있을 경우 가장 먼저 나타나는 부분을 매칭 부분으로 규칙이다. 그리고 Greedy Rule이란 패턴에 매칭되는 부분을 가능한 한 많이 포함시킨다는 규칙이다. 다음의 예에서 실제로 매칭된 부분을 보면서 이 2가지 규칙의 의미를 음미해 보기 바란다.

	패턴
	a+

	매치되는 문자열
	aa is aaaaaa and aaaa

aaaaaa12 or aaaaaaaaaaaa13

AAAaaaaaaaaaa+ 1243 aaa


1.6.2. Backslash Escape

자주 사용되는 패턴을 미리 정의하여 간편하게 이용할 수 있도록 한 기능이다. 이 기능은 Tcl/Tk 8.1 이상에서만 사용 가능하다는 점에 유의하기 바란다. 제공되는 Backslash Escape 중에 자주 사용될 수 있는 것들은 다음과 같다. **표시 된 것들은 Tcl/Tk 8.0 이하 버전에서도 제공되던 것들이다. 모두 글자 하나를 의미함을 명심하자.

	Backslash Escape
	의미

	\d
	숫자

	\D
	숫자가 아닌 모든 것

	\s
	공백 문자 ( 스페이스 혹은 탭 )

	\S
	공백 문자가 아닌 모든 것

	\w
	워드 구성 성분 중 하나 ( 알파벳과 숫자 혹은 밑줄 문자 )

	\W
	워드 구성 성분이 아닌 모든 것

	\n
	개행 문자 ( new line ) **

	\t
	탭 문자 ( tab ) **

	\a
	Alert

	\b
	backspace

	\e
	ESC 키


그림 2 주요 Backslash Escapes

Backslash Escape를 사용하면 한 단어를 찾는 패턴을 다음과 같이 간단히 작성할 수 있다.

	패턴
	\w+

	매치되는 문자열
	a123_a is aaaaaa and aaaa

+!- 12_b_34_* 1243 aaa


1.6.3. regexp 명령어
regexp 명령어는 정규 표현식에 주어진 문자열이 매칭될 때 1을 리턴하고 매칭되지 않으면 0을 리턴한다. 그리고 원하는 문자열을 지정한 변수에 저장할 수 있는 방법을 제공한다. regexp의 정확한 문법은 다음과 같다.

regexp ?flags? expression string ?allmatch variable1 variable2 … ?

expression에는 정규 표현식이 오고 string에는 정규 표현식으로 매칭 여부를 판별할 문자열이 오는 자리이다. 여기까지만 명시하면 매칭 여부만을 알 수 있다. 

만일 그 다음에 변수를 지정하면 첫번째 지정된 변수에는 매칭되는 패턴 전체가 복사된다. 그리고 그 다음 변수에는 사용자가 “( )”를 써서 매칭된 패턴 중에서 특정한 부분을 지정했을 경우 차례대로 저장된다. 다음에 이에 대한 예제가 있다. 여기서 주의할 점은 정규 표현식의 가독성(readability)을 높이기 위하여 공백문자를 임의로 삽입해서는 안 된다는 점이다. 공백 문자도 엄연히 하나의 패턴으로 작용하기 때문에 원치 않는 결과를 초래하게 된다.

regexp명령어에 적용 가능한 flag들은 다음과 같다.

*  -nocase : 정규 표현식에 나타난 소문자들은 문자열의 같은 문자이기만 하면 소문자와도 대문자와도 매칭된다.

*  -indices : 문자열 안에서 매칭된 패턴의 시작 인덱스와 끝 인덱스의 쌍을 주어진 변수에 기억 시킨다. 문자열 인덱스는 0부터 시작한다.

*  -expaned : 패턴을 여러 줄에 걸쳐서 기술할 수 있도록 해준다. 그리고 각 패턴에 대하여 주석을 달 수 있도록 해준다.

*  -- : 정규표현식이 “ - ”로 시작할 경우 이 플래그를 적용하여 플래그로 오해할 수 있는 소지를 없앤다.

	% set test_string host:address:phone:name

host:address:phone:name

% regexp {([a-z]+):([a-z]+):([a-z]+)} $test_string all p1 p2 p3

1

%set all

host:address:phone

% set p1

host

% set p2

address

% set p3

phone

% regexp –indices {([a-z]+):([a-z]+):([a-z]+)} $test_string all p1

1

% set all

0 17

% set p1

0 3


위에서 보는 바와 같이 regexp를 사용하여 문자열에서 원하는 부분을 뽑아낼 수 있다. 뽑아내고자 하는 패턴을 “( )”로 감싸고 난 후 “( )”의 순서대로 변수의 이름을 배열하면 된다. 단, 여기서 주의할 점은 변수 all의 값에서 보는 바와 같이 첫번째 변수에는 매칭된 모든 부분이 저장된다는 것이다.

regexp를 사용할 때 잦은 실수 중의 하나는 다음과 같은 경우이다.

	% set string yahoo

yahoo

% regexp { yahoo } $string

0


위의 예제에서 regexp의 실행 결과가 1이 나와야 하는데 0이라고 이상하게 생각하는 사람이 많다. 여기서 0이 나올 수 밖에 없는 이유는 매칭 패턴에 공백 문자가 포함되어 있기 때문이다. 따라서 다음과 같이 고쳐야만 원하는 결과를 얻을 수 있다. 

	% set string yahoo

yahoo

% regexp {yahoo} $string

1


이 예에서 보는 바와 같이 공백문자도 패턴이라는 점을 명심하여야 한다. 정규표현식의 패턴을 쓸 때는 반드시 필요한 패턴만을 써야한 다는 점을 잊지 말아야 한다.

-expaned 옵션을 사용하면 다음과 같이 패턴에 대한 주석을 달 수가 있다. 보통의 경우 정규 표현식은 매우 복잡하여 해석하기가 매우 곤란한 경우가 많은데 이 옵션을 이용하면 이러한 문제를 극복할 수 있다. 다음의 두가지 명령어는 똑같은 기능을 수행한다. -expaned옵션을 사용했을 때 어떤 장점이 있는지 눈으로 확인해 보기 바란다. -expanded옵션을 사용했을 때 패턴내에 삽입된 모든 공백 문자와 개행문자 그리고 주석은 무시된다. -expanded옵션을 사용했을 때 공백문자나 개행문자를 패턴에 포함 시키고자 할 때는 backslash escape중 \s와 \n을 사용해야 한다.

	regexp {[+-]?\d*\.\d+} $string matchpart

	regexp -expanded { [+-]?      # 실수의 부호가 나와도 좋고 안나와도 좋고
                       \d*        # 실수의 정수부분은 있어도 좋고 없어도 좋고
                       \.         # 소수점은 반드시 있어야 되고
                       \d+        # 실수에서 소수부분은 반드시 있어야 된다.

                     } $string matchpart


1.6.4. regsub 명령어
regsub는 정규 표현식을 이용하여 치환 연산 (substitution)을 수행하도록 하는 명령어이다. 이 명령어의 정확한 문법은 다음과 같다.

regsub ?flags? expression string subspec variable
regexp에서와 마찬가지로 expression과 string에는 각각 정규 표현식과 매칭 여부를 검사할 문자열이 온다. 다음에 어떻게 치환을 행할 것인지를 명시하는 subspec이 온다. 그리고 마지막으로 치환 결과를 저장할 변수가 온다. 이 명령은 수행 후 치환이 일어난 개수를 리턴한다.

regsub에 적용 가능한 플래그들은 다음과 같다.

*  -nocase : 정규 표현식의 소문자는 같은 문자라면 문자열의 대문자, 소문자 모두와 매칭 된다.

*  -all : 매칭이 여러 번 일어나면 모두 치환 하라
*  -- : 정규 표현식이 “ – ” 문자로 시작할 경우 플래그로 오해하지 않도록 하기 위한 플래그이다.

	% set test_string host:address:phone:name

host:address:phone:name

% regsub {([a-z]+):([a-z]+)} $test_string {\1 + \2} sub_string

1

% set sub_string

host + address:phone:name

% set test_string everybody

% regsub {e} $test_string {E} sub_string

1

% set sub_string

Everybody

% regsub –all {e} $test_string {E} sub_string

2

% set sub_string

EvErybody


regexp에서 처럼 뽑아내고 싶은 부분이 있으면 이 부분을 “( )”로 감싸준다. “( )”의 순서에 따라 각 괄호안의 패턴에 매칭된 부분은 \1, \2, \3 ... 등의 변수에 저장된다. 이를 이용하여 치환 패턴에 사용할 수 있다. 단, 여기서 주의할 점은 \1, \2, \3 ... 등과 같은 변수는 치환 패턴 안에서만 사용할 수 있고 다른 곳에서는 사용할 수 없다는 점이다.

regsub명령어에서 정규 표현식 패턴을 쓸 때 유의할 점은 “\t”가 Tab 문자로 매칭이 되지 않는 다는 점이다. 현재 버전의 버그인지 다른 이유로 인한 것인지는 알 수 없다. Tab 문자를 패턴에 넣기 위해서는 실제로 Tab 문자를 치는 수 밖에 없다.

연 습 문 제
1. 다음과 같은 데이타를 입력 받아 출력 결과와 같이 변환하시오. 데이타의 입력은 데이타 분석 연습 문제에서와 같은 방법으로 하시오.

입력 데이타
	DAT_LOT3

DAT_LOT1

DTT_PHDA4

HHUI_EE2


출력 결과
	LOT-3DAT

LOT-1DAT

PHDA-4DTT

EE-2HHUI


2. 다음의 데이타 파일은 간략화된 Verilog 디자인 파일이다. 이를 분석하여 입출력의 종류를 출력하는 프로그램을 작성하시오. 데이타 파일의 입력은 데이타 분석 연습문제에서와 같은 방법으로 하시오.

입력 데이타
	module  aa(a, b, c, d);

input a, b;

output c;

inout d;

....;

endmodule;

module bb(a1, a2, a3, a4, a5, a6, a7,a8);

input a1;

input a2;

input a3;

output a4,

       a5,

       a6 ;

inout a7;

inout a8;

....

endmodule;


출력 결과
	MODULE NAME : aa

    inputs : a b

    outputs : c

    inouts : d

MODULE NAME : bb

    inputs : a1 a2 a3

    outputs : a4 a5 a6

    inouts : a7 a8


1.7. 함수
Tcl에서 함수를 정의하고 사용하는 방법을 알아본다. 그리고 함수 내에서 사용하는 변수의 참조 범위와 관련된 여러 가지 사항들에 관하여 알아 본다.

1.7.1. 함수의 정의
함수를 정의하는 문법은 다음과 같다.

proc  procName parameterNames procBody
예를 보이면 다음과 같다. 여기서 주의할 점은 함수를 사용하기 전에 반드시 먼저 함수를 정의해야 한다는 점이다. 다시 말해서 함수 정의 부분이 파일의 앞 쪽에 오고 파일의 뒤 쪽에서 함수 호출 부분이 있어야 한다
.

	proc factorial { a } {

     set result 1

     for { set t 1 } { $t <= $a } { incr t } {

         set result [ expr $result * $t ]

     }

     return $result

}


1.7.2. 함수의 사용
함수를 사용하는 방법은 일반 명령어를 사용하는 방법과 같다. 보는 바와 같이 함수에 넘겨 줄 Argument가 있을 경우 이를 공백문자를 사이에 두고 나열해 주면 된다. 공백이 포함된 문자열을 함수의 Argument로 넘겨줄 때에는 “ ”로 감싸 주면 된다.

	% factorial 4

24


1.7.3. Default Parameter Value

함수의 Argument을 미리 정해 놓고 함수 호출 시 Argument의 값을 명시하지 않는 경우에는 미리 정해진 값을 사용하도록 할 수 있다.

	% proc yahoo { {a 2} {b 3} {c 4} } {

     puts [expr $a + $b + $c]

}

% yahoo

9

%  yahoo 4

11

% yahoo 4 5

13

% yahoo 4 5 6

15

% yahoo 4 5 6 7

called "yahoo" with too many arguments


1.7.4. Parameter갯수의 유동성
일반적인 경우에 함수의 Parameter갯수는 일정하게 정해져 있다. 따라서 함수 호출시 정해진 Parameter의 갯수보다 많은 인자를 넘겨줄 떄는 다음과 같이 에러가 발생한다.

	% proc haha { a } {

    puts $a

}

% haha 1

1

% haha 1 2 3

called "haha" with too many arguments

% 


그러나 어떤 경우에는 함수의 Parameter 갯수를 명확히 정하기 힘든 경우가 있다. 이런 경우 args라는 특수한 Parameter를 이용하여 다음과 같이 프로그램할 수 있다. args라는 Parameter가 존재하는 경우는 함수에 정의 된 Parameter의 갯수보다 많은 인자가 넘어올 경우 나머지는 모두 args라는 변수에 리스트 형태로 저장된다. args는 반드시 맨 마지막에 와야 한다. 

	% proc hoho { a args } {

    puts “a : $a”

    puts “args : $args”

}

% hoho

no value given for parameter "a" to "hoho"

% hoho 1

a : 1

args :

% hoho 1 2

a : 1

args : 2

% hoho 1 2 3 4 5 6

a : 1

args : 2 3 4 5 6


1.7.5. 변수 참조 범위
Tcl에서는 특별한 선언을 하지 않은 변수들은 모두 local변수로 만들어 진다. local변수란 함수가 실행되는 동안만 존재하고 함수의 실행이 끝나면 사라지는 변수를 말한다. Global 변수를 함수내에서 참조하기 위해서는 반드시 사용하기 전에 global변수임을 다음과 같이 명시하여야 한다.

	proc mySub { } {

     global a       ; #변수 a는 global 변수임을 선언
     set a 5       

     set b 5        ; #변수 b에 대한 특별한 선언이 없으므로 b는 local변수로 생성됨
}


Tcl함수 내에 존재하는 변수는 이름과 상관 없이 독립적으로 존재한다. 따라서 여러 함수 안에서 같은 이름의 변수를 서로 간섭 받지 않고 사용할 수 있다. 그러나 같은 함수 내에서는 같은 이름의 변수는 같은 변수로 본다. 다음의 예에서 보는 바와 같이 global 변수와 같은 이름의 local 변수가 사용되는 경우 global 변수는 함수내에서 보이지 않게 된다. 

	% set a 5

5

%  proc ttt {} {

       set a 2

       puts $a

}

% proc kkk {} {

       set a 4

       ttt

       puts $a

}

% kkk

2

4

% set a

5


Tcl함수의 매개변수로 변수가 주어졌을 때 변수 자체를 넘겨 주지 않고 변수의 값만을 함수에게 넘겨 준다. 따라서 함수 내에서 함수 Argument의 값을 변경하여도 매개변수로 주어진 변수의 값에는 변함이 없다.
 변수의 값을 바꾸어 주는 함수를 작성하기 위해서는 upvar명령어를 사용하여야 한다.

	% set a 5

5

%  proc ttt { a } {

       set a 2

}

% ttt a

2

% set a

5


1.7.5.1. 전역 변수 (Global Variable)

전역 변수를 함수 안에서 사용하기 위해서는 global이라는 명령어를 사용한다. global이라는 명령어 다음에 주어지는 변수들은 모두 전역 변수라는 것을 Tcl인터프리터에 알린다. 전역 변수는 어떤 함수에서도 같은 이름으로 참조 가능 하다.

다음의 예제를 음미하면서 전역 변수 사용법을 이해하도록 한다.

	% set a 10

10

% set b 20

20

% proc ttt {} {

       puts “$a $b”

}

% ttt

can't read "a": no such variable

% proc hhh {} {

        global a b

        puts “$a $b”

}

% hhh

10 20

% proc kkk {} {

    global a

    set a 5

}

% kkk

% set a

5


버전 8.0이전 버전에서는 다음과 같이 배열을 전역 변수로 이용하기 위해서 global 다음에 array를 써서 배열임을 명시해야 한다.
 그러나, 버전 8.0 이후 부터는 배열 변수에 대해서도 array라는 키워드를 명시하지 않아도 되도록 바뀌었다.

	% set TABLE(“my name”) nara

nara

%  proc ttt {} {

       # 버전 8.0이전에는 다음과 같이 array임을 명시해야 했음
       global array TABLE

       puts $TABLE(“my name”)

}

% ttt

nara


1.7.5.2. upvar

upvar명령어는 변수의 참조 범위를 변경시켜주는 명령어이다. upvar명령어의 사용 형식은 다음과 같다.

upvar ?stackLevel? varName localVar

upvar명령어를 이용하여 Call by Name형식의 매개변수를 구현할 수 있다. Call by Name이란 함수의 매개변수가 변수일 때 변수 자체를 함수로 넘겨 주는 방식을 말한다. Call by Name방식에서는 함수 내에서 매개변수로 넘어온 변수의 값을 바꿀 수 있다. 다음에 그 예제가 있다. 

	#!/usr/local/bin/tclsh

proc MyIncr { varName } {

    upvar $varName a

    set a [ expr $a + 1 ]

}

set a 5

MyIncr a ; #MyIncr will change the value of a

puts $a




위에서 MyIncr을 호출할 때 변수명 앞에 “$” 기호가 없을 주의하자. 서브루틴으로 변수의 값을 넘기는 것이 아니라 변수의 이름을 넘기는 것이다. 이러한 서브루틴 호출 방식을 Call by Name이라고 한다. 서브루틴으로 넘겨진 변수 이름을 가지고 메모리 상에 존재하는 변수값 저장 위치를 찾아내어 그 변수의 값을 바꾸거나 값을 참조할 수 있도록 해주는 명령어가 바로 upvar인 것이다.

위와 같은 프로그램을 작성하라고 했을 때 보통 다음과 같은 실수를 하기 쉽기때문에 각별한 주의를 해야 한다.

	proc MyIncr { varName } {

    # This is not valid programming

    # below statement has no effects

    set varName [ expr $varName + 1 ]

}


함수가 호출되면 스택에 그 함수에서 사용할 지역 변수 (local variable)들을 저장하게 된다. 따라서 함수가 함수를 부르고 그 함수에서 또 함수를 부르고 이런 식으로 반복되면 점점 스택의 깊이는 깊어지게 된다. upvar명령어에서 stackLevel이라는 것은 이것을 말한다.

staclLevel이 0이면 바로 자신의 스택을 말하고 1이면 자신을 호출한 함수의 스택을 말한다. 다음 예제를 통하여 그 의미를 이해하도록 하자.

	% proc p1 {} {

    upvar 1 a a1

    upvar 2 a a2

    upvar 3 a a3

      set a 1

    puts "a0 = $a"

    puts "a1 = $a1"

    puts "a2 = $a2"

    puts "a3 = $a3"

}

% proc p2 {} {

    set a 2

    p1

}

% proc p3 {} {

   set a 3

   p2

}

% set a 4

4

% p3

a0 = 1

a1 = 2

a2 = 3

a3 = 4

%


위의 예제를 실행 시켰을 때 서브루틴 p1이 실행되고 있는 순간 스택의 모양을 간단하게 그려 보면 다음과 같다. 각 함수의 실행이 끝나면 각 함수가 stack상에서 차지하고 있던 공간은 사라진다. 





그림 3 stack의 snap shot

전역 변수들이 존재하는 스택을 가리키는 특별한 코드는 “#0”이다.

	% proc p1 {} {

    upvar #0 a aglobal

    puts "a = $aglobal"

}

% proc p2 {} {

    p1

}

% proc p3 {} {

   p2

}

% set a 4

4 

% p3

a = 4

%


stackLevel을 0으로 하면 변수 별명(Variable Alias)을 만들 수 있다.

	% proc ttt {} {

      set a 5

    upvar 0 a b

    set b 1

    puts "a = $a"

}

% ttt

1


1.8. 파일 및 디렉토리 조작
스크립트 프로그래밍에서 가장 많이 사용되는 것이 파일을 열고 그 파일의 내용을 읽거나 그 파일에 결과를 출력하는 것일 것이다. 또한 파일과 디렉토리를 생성, 복사, 이동하고 그 속성(Attribute)을 바꾸는 일도 매우 자주 한다. Tcl/Tk는 이러한 일들을 수행할 수 있는 강력한 기능을 제공한다. 또한 일반적인 파일 뿐 아니라 Pipe까지도 파일처럼 사용할 수 있도록 해주고 있다.

1.8.1. OS마다 다른 파일 경로 표기법
파일의 경로를 표현하기 위해 사용되는 표기법이 OS마다 제각각이다. 하지만 Tcl/Tk에서는 모든 파일명을 UNIX스타일로 사용하면 Tcl/Tk인터프리터가 현재 사용하고 있는 OS에 맞게 자동적으로 변환해 준다. 따라서 스크립트 작성시 Relative Path Name을 이용하면 파일 경로명을 OS에 따라 달리 줄 필요가 없어 진다. 

	% dir d:/wonnr/tcl

 드라이브 D에 이름이 없습니다
 일련 번호 17FE-0C66

 디렉토리 D:\wonnr\TCL\교재\예제
.              <DIR>        98-09-29  10:53 .

..             <DIR>        98-09-29  10:53 ..

TREE     TCL           978  98-09-19   9:12 tree.tcl

MOVE     TCL           592  98-09-28  10:47 move.tcl

RES      TCL           201  98-09-28  15:36 res.tcl

TEST     RES           313  98-09-28  15:35 test.res

MENU     TCL         2,665  98-09-28  13:52 menu.tcl

TEST     TCL            19  98-09-29  14:13 test.tcl

         6개 파일                4,768 바이트
         2개 디렉토리      606,081,024 바이트 사용가능
%


그러나 Path Name을 다른 프로그램의 Argument로 넘겨 줘야 할 경우에는 각 OS에 따라 OS가 정하는 규칙에 맞는 경로명을 만들어 줘야 한다. 이러한 경우를 대비하여 Tcl/Tk에서는 file join명령어와 file split명령어를 제공하고 있다.

1.8.2. 파일 입출력
파일로부터 자료를 입력받거나 자료를 출력하기 위해서는 다음과 같은 3가지 단계를 거쳐야 한다.

파일 열기
파일로 데이터를 입출력
파일 닫기
1.8.2.1. 파일 열기
파일을 열기 위해서 사용하는 명령어는 open 명령어이다. open명령어는 파일을 연 후 그 파일에 대한 핸들러를 리턴한다.

open fileName ?accessType? ?permissions?

fileName에는 열고자 하는 파일의 이름이나 경로명을 주면 된다. accessType에 가능한 값은 표 10에 정리하였다. 사용할 수 있는 값의 종류는 간략형과 POSIX형으로 나눌 수 있는데 이 둘을 동시에 같이 사용할 수는 없다. 간략형은 C언어의 라이브러리 함수인 fopen()에서 사용하는 값과 의미가 같고, POSIX형은 open()함수에서 사용하는 값과 의미가 같다. POSIX형은 리스트를 사용하여 여러 값을 조합할 수 있다.

	set dataFileIN [ open ../DATA/result01.dat r ]

set dataFileOUT [ open ../DATA/result02.dat { RDWR CREATE } ]


파일을 열 때는 항상 예상치 않은 오류가 발생할 수 있다는 점에 유의하여야 한다. 이에 대비하여 catch 구문을 이용 해 다음과 같은 조치를 취하여 주는 것이 좋다.

	if [ catch { set fileID [ open “my.dat” r ] } errorMessage ] {

    puts stderr “Error : $errorMessage”

} 

...

close $fileID


이 경우 Open된 File의 핸들러는 fileID라는 변수에 저장이 되고 File Open시 에러가 발생하면 그에 대한 에러 메시지가 errorMessage라는 변수에 저장되게 된다. 위와 같은 코드는 File Open의 Idiom이라고 할 수 있으므로 꼭 익혀 두도록 하자. 보다 간단하게 다음과 같이 프로그램할 수도 있다.

	if [ catch { open “my.dat” r } fileID ] {

    puts stderr “Error : $fileID”

} else {

    ...

    close $fileID

}




이 경우에는 에러 없이 파일이 열린 경우에는 파일 핸들러가 fileID에 저장되고 에러가 발생한 경우에는 fileID라는 변수에 에러 메시지가 저장된다. 따라서 위와 같이 에러가 발생한 경우와 발생하지 않은 경우 fileID의 내용이 다르므로 사용에 유의하여야 한다.

	Access Type
	의미

	간
략
형
	r
	읽기 전용. 파일은 반드시 존재해야 함

	
	r+
	읽고 쓰기 가능. 파일은 반드시 존재해야 함

	
	w
	쓰기 전용. 파일이 이미 존재하면 덮어 쓴다.

	
	w+
	읽고 쓰기 가능. 파일이 이미 존재하면 덮어 쓴다.

	
	a
	쓰기전용. 존재하는 파일의 내용이 보존됨. 파일이 반드시 존재해야 함

	
	a+
	읽고 쓰기. 존재하는 파일의 내용이 보존됨. 파일이 반드시 존재해야 함

	P

O

S

I

X

형
	RDONLY
	읽기 전용

	
	WRONLY
	쓰기 전용

	
	RDWR
	읽기 쓰기

	
	APPEND
	기존 파일 내용 보존

	
	CREATE
	파일이 없다면 새로 만든다.

	
	EXCL
	CREATE과 함께 사용될 경우, 이미 파일 존재하면 에러를 발생

	
	NOCTTY
	파일명이 terminal device인 경우 terminal 제어를 허용하지 않는다.

	
	NONBLOCK
	파일을 여는 모든 과정이 끝날 때 까지 스크립트의 실행을 멈추고 있는 것을 방지한다. 파일을 여는 과정이 끝나지 않았더라도 스크립트의 수행은 계속 된다.

	
	TRUNC
	파일이 이미 존재하면 내용을 삭제하고 길이를 0으로 만든다.


표 10 open access types
1.8.2.2. 파일 닫기
파일을 닫기 위해 사용되는 명령어는 close이다. open명령어가 리턴한 파일 핸들러를 주면 된다. 

	set dataFileIN [ open ../DATA/result01.dat r ]

close $dataFileIn


파일을 열고 close를 이용하여 파일을 닫지 않았을 경우 스크립트가 종료되면 자동으로 파일은 닫힌다. 하지만 close를 하지 않은 상태에서 많은 파일을 계속 열 경우 시스템 자원이 고갈되어 에러가 발생할 가능성이 있다.

1.8.2.3. 파일 쓰기
파일에 데이터를 쓰는 명령에는 puts명령어가 있다. puts명령어는 주어진 문자열과 새줄문자( new line character )를 출력한다. 

puts ?-nonewline? ?channelID? stringToWrite

channelID에는 open명령어가 리턴한 파일 핸들러 값을 주면 된다. channelID를 명시하지 않으면 기본적으로 puts는 stdout으로 출력한다. stdout은 항상 열려 있으므로 따로 open할 필요는 없다. stdout외에 stdin, stderr가 기본적으로 항상 열려 있다. stdout과 stderr는 출력 채널이며 stdin은 입력 채널이다.

기본적으로 출력되는 새줄문자가 출력되지 않게 하기 위해서 -nonewline옵션을 사용할 수 있다.

	puts “Program Started”

set handle [open MyFileName w]

puts $handle “This line will be printed in MyFileName”

close $handle

puts “Program was over”


1.8.2.4. flush명령어
파일에 출력을 실행할 때에는 기본적으로 버퍼에 관하여 동작하기 때문에 프로그램이 계속 동작 중인 동안에는 출력 파일을 열어 봐도 아무런 내용을 찾을 수 없는 경우가 대부분이다. 따라서 만일 서로 다른 프로그램이 같은 파일을 공유하면서 서로 데이타를 주고 받고자 할 경우 파일에 출력하는 프로그램이 완전히 끝나야지만 다른 프로그램이 데이타를 받아갈 수 있게 되는 문제점이 발생한다.

이러한 경우에 즉시 파일에 출력하기 위하여 flush라는 명령어를 호출하면 출력하는 즉시 실제 파일에 출력 내용을 저장할 수 있다.

	set handle [open MyFileName w]

puts $handle “This line will be printed in MyFileName”

flush $handle
close $handle

puts “Program was over”


1.8.2.5. 파일 읽기
파일을 읽는 명령어는 gets와 read가 있다. 

gets ?channelID? ?variableName?

gets명령어는 한 번에 한 줄씩 명시된 채널에서 읽어 들인다. channelID를 명시하지 않으면 기본적으로 stdin에서 한 줄을 읽어 들인다. 변수 이름을 variableName에 명시하면 읽어들인 문자열은 변수에 저장하고 읽어들인 문자의 갯수를 리턴한다. 변수 이름을 명시하지 않으면 읽어들인 문자열을 리턴한다.

read ?-nonewline? channelID

read channelID nuberOfByte

read명령어는 명시된 채널의 전체 내용을 읽어 들이거나 주어진 바이트 수 만큼의 내용을 읽어들인다. -nonewline 옵션이 주어지면 마지막 글자는 새줄 문자로 간주하고 버린다. read명령어는 읽어들인 내용을 리턴한다.

다음에 gets와 read명령을 이용하여 “/etc/passwd”파일의 내용을 출력하는 프로그램이 있다.

	#!/usr/local/bin/tclsh

set input [open /etc/passwd r]

while { [gets $input line] >= 0 } {

    puts $line

}

close $input


	#!/usr/local/bin/tclsh

set input [open /etc/passwd r]

puts [read $input]

close $input


1.8.2.6. 파이프 열기
Tcl/Tk에서는 device파일을 포함한 일반적인 파일 뿐 아니라 Pipe도 open명령을 통하여 열 수 있다. 파일명을 “ | ”로 시작하면 파이프로 인식한다.

다음 프로그램은 “sort /etc/passwd”의 결과를 출력하는 프로그램이다. Pipe를 읽기 전용으로 열어서 사용하고 있다.

	#!/usr/local/bin/tclsh

set inputData [open “| sort /etc/passwd” r]

puts [read $inputData]

close $inputData


다음 프로그램은 “/etc/passwd”를 UNIX명령어인 “sort”를 이용해 정렬하는 프로그램이다. 위의 프로그램과 다른 점은 Pipe를 쓰기 전용으로 열었다는 데에 있다.

	#!/usr/local/bin/tclsh

set input [open /etc/passwd r]

set output [open “| sort” w]

puts $output [read $input]

close $input

close $output


1.8.3. file명령어
Tcl/Tk에서는 파일과 관련된 많은 기능들을 file명령어를 통하여 제공한다. file명령어의 여러 가지 사용법을 표 11에 정리하였다.

	문법
	의미

	file atime name
	name이 최종적으로 access된 시간을 정수로 리턴한다. 이 정수는 1970년 1월 1일부터 경과된 시간을 초로 환산한 시간이다.

	file attributes name ?option?
	파일 속성에 대한 여러 가지 값을 리턴한다.

	file copy ?-force? src dest
	src를 dest로 복사한다. -force옵션은 dest가 이미 존재하는 경우 덮어쓰도록 한다.

	file delete ?-force? name
	name을 삭제 한다. -force옵션은 비어 있지 않은 디렉토리도 삭제할 수 있도록 해준다.

	file dirname name
	name의 문자열을 분석하여 디렉토리 path에 해당하는 부분만을 추출하여 리턴한다. 디렉토리 path에 해당하는 부분이 없으면 “ . ”을 리턴한다.

	file executable name
	실행 가능한 파일일 경우 1을 리턴한다.

	file exists name 
	name이라는 이름을 가진 파일이 존재하는 경우 1을 리턴한다.

	file extension name
	name의 문자열 중 마지막 구두점 이후의 문자열을 리턴한다. 구두점이 없으면 빈 문자열을 리턴한다.

	file isdirectory name
	name이 디렉토리이면 1을 리턴한다.

	file join path path ...
	각 path들을 하나의 path로 합친다. path를 합칠 때 사용되는 경로 분리자(path separator)는 각 OS에 알맞은 것을 사용 한다.

	file lstat name var
	symbolic link파일이 link하고 있는 파일에 대하여 file stat기능을 수행한다.

	file mkdir name
	name이라는 디렉토리를 만든다.

	file mtime name
	name이 마지막으로 수정된 시간을 정수로 리턴한다. 이 정수는 1970년 1월 1일부터 경과된 시간을 초로 환산한 시간이다.

	file nativename name
	name에 주어진 경로명을 각 OS에 알맞는 형태로 바꾸어 리턴한다.

	file owned name
	name의 소유자가 현재 사용자일 때 1을 리턴한다.

	file pathtype name
	absolute, relative, volumerelative중에 하나를 리턴한다. Absoute은 name이 절대 경로 일때, relative는 상대 경로일때 리턴한다. voluemrelative는 “C:\wonnr”과 같이 volume이 명시된 경우에 리턴한다.

	file readable name
	name을 읽을 수 있는 권한이 있으면 1을 리턴한다.

	file readlink name
	symbolic link가 가리키고 있는 파일의 이름을 리턴한다.

	file rename ?-force? old new
	파일 이름을 바꾼다. -force옵션은 new에 명시된 이름과 같은 파일이 있는 경우 덮어 쓴다.

	file rootname name
	name의 문자열이 구두점으로 구분되어 있을 때 구두점 왼쪽의 모든 값을 리턴한다. 구두점이 없을 경우에는 name의 문자열을 그대로 리턴한다.

	file size name
	파일의 크기를 바이트수로 리턴한다.

	file split name
	name에 주어진 경로명에서 경로 분리자(path separator)로 나뉘어진 요소들로 분리하여 리스트로 만든 후 그 리스트를 리턴한다. name에 주어 지는 문자열에는 각 OS에 알맞는 경로명을 사용해도 된다.

	file stat name var
	Kernel system call인 stat을 호출하여 결과로 리턴되는 파일에 관한 정보를 var에 저장해 준다.

	file tail name
	name에 주어진 경로명 중에서 맨 마지막 경로 분리자 (path separator)다음에 있는 문자열을 리턴한다.

	file type name
	파일의 type을 검사하여 다음 값 중에 하나를 리턴 한다. file, directory, characterSpecial, blockSpecial, fifo, link, socket

	file volume
	현재 시스템에 사용 가능한 volume들을 리스트로 만들어 리턴한다. (예 : { a:/ c:/ } )

	file writable name
	쓸 수 있는 권한이 있으면 1을 리턴 한다.


표 11 file명령어 사용법
1.8.4. cd와 pwd 명령어
UNIX명령어와 마찬가지로 cd는 현재 디렉토리를 다른 곳으로 옮기는 기능을 하고 pwd는 현재 작업 디렉토리의 경로명을 리턴한다.

연 습 문 제
1. 현재 디렉토리 내에 존재하는 모든 파일들의 라인수의 합을 구하여 출력하는 프로그램을 작성하시오. 이 프로그램의 결과는 UNIX명령어 wc -l  * 를 실행한 것과 같은 결과임
1.9. 외부 프로그램과의 연동
Tcl 프로그램 내에서 외부 프로그램을 호출하는 방법과 Tcl프로그램의 수행 종료 시 종료 상태를 외부 프로그램에게 알리는 방법에 대하여 알아 본다.

1.9.1. exec 명령어
exec명령어는 외부 프로그램을 Tcl스크립트 안에서 실행시켜 준다.  

exec ?-keepnewline? uinixCommand

exec는 기본적으로 명령이 한 줄로 끝난다고 생각한다. 그러나 -keepnewline을 명시하면 여러 줄에 걸친 외부 명령을 exec를 이용하여 실행할 수 있다. 외부 명령어인 ls -al을 수행하고 그 결과를 출력하는 스크립트는 다음과 같다. Windows 95/NT인 경우는 ls대신 dir명령을 실행해 보도록 한다.

	puts [exec ls -al]; # for UNIX


	puts [exec dir]; # for Windows 95/NT


외부 프로그램이 실행되다가 에러가 발생한 경우 에러 메시지를 화면에 찍히지 않게 하고 스크립트 내부에서 처리하고자 한다면 다음과 같이 프로그램 하면 된다. 

	if { [ catch { exec haha } errMessage ] } {

   puts "Error : $errMessage"

} else {

   puts “Result : $errMessage”

}


위와 같이 프로그램하면 외부 명령어가 stdout으로 출력하는 결과와 stderr로 출력하는 결과가 모두 errMessage에 저장되게 된다. stdout으로 출력되는 결과와 stderr로 출력되는 결과를 분리하고 싶은 경우 다음과 같이 프로그램하면 된다.

	if { [ catch { set result [ exec haha ] } errMessage ] } {

   puts "Error : $errMessage"

} else {

   puts “Result : $result”

}


위와 같이 프로그램하면 stdout으로 출력되는 메시지는 result라는 변수에 저장되고 stderr로 출력되는 메시지는 errMessage라는 변수에 저장되게 된다.

만일 exec로 실행하려는 프로그램이 Tcl로 작성한 프로그램일 경우 위와 같이 catch가 제대로 동작하기 위해서는 다음과 같은 조건을 만족해야 한다. 정상적으로 수행을 마쳤을 때는 Command Shell
에 종료코드로 0을 리턴하고 에러가 발생한 경우는 0이 아닌 값을 리턴 한다.
 그리고 모든 에러 메시지는 stderr로 출력한다. 종료 코드의 값은 exit 명령어의 Argument로 주면 된다.

exec에서도 UNIX에서 제공하는 IO redirection기능을 모두 사용할 수 있다. 다음에 그 예가 있다.

	% exec sort < myDataFile | uniq >> mySortedData


exec를 통하여 외부 명령을 수행하면 외부 명령이 끝날 때까지 기다린다. 이를 방지하기 위하여 명령어 맨 끝에 “ & ”를 붙여 외부명령을 백그라운드로 수행시킬 수 있다. 이 때 exec는 process ID를 리턴한다.

1.9.2. exit 명령어
exit ?exitCode?
exit 명령어는  Tcl쉘을 종료한다. Argument을 지정하지 않으면 정상적인 종료 코드인 0을 command shell에 리턴한다. 매개 변수 값을 지정하여 원하는 종료 코드를 리턴할 수 있다. exit을 사용하지 않고 Tcl쉘이 종료된 경우에는 정상적인 종료 코드를 리턴한다.

	if { ... } {

    # error case

    exit 1

}


스크립트가 정상적으로 종료하는 경우에는 종료코드 0을 리턴하고 에러가 발생한 경우에는 0이 아닌 값을 종료 코드로 리턴하여야 한다.

1.10. 기타
앞의 장에서 언급되지 않았지만 중요한 내용들에 대하여 간략히 기술하였다.

1.10.1. 명령행 인수 (Command Line Argument)

Tcl 프로그램에서 명령행 인수는 argv 라는 리스트에 차례대로 저장되어 있다. C언어에서는 argv의 첫번째 원소는 프로그램의 이름이 저장되어 있으나 Tcl에서는 첫번째 인수가 저장되어 있다. 프로그램의 이름을 알고 싶으면 argv0라는 전역 변수를 참조 하면 된다. 인수의 갯수는 argc라는 변수에 저장되어 있다.

다음을 실행시켜 보면서 그 의미를 확실히 이해 하도록 하자.

	#!/usr/local/bin/tclsh

puts “Program name is : $argv0”

puts “Number of arguments : $argc”

foreach  arg [ $argv ] {

    puts “Command Line Argument : $arg”

}


이 프로그램을 comarg.tcl이라는 파일로 저장하였다면 UNIX 쉘이나 DOSBOX안에서 다음과 같이 실행하면 다음과 같은 결과가 나온다.

	$ comarg.tcl haha hehe hoho

Program name is : comarg.tcl

Number of arguments : 3

Command Line Argument : haha

Command Line Argument : hehe

Command Line Argument : hoho

$ comarg.tcl 1 2

Program name is : comarg.tcl

Number of arguments : 2

Command Line Argument : 1

Command Line Argument : 2

$


1.10.2. 환경 변수 (environment Variable)

Command Shell에는 여러 프로그램이 정보를 공유할 수 있도록 해주는 환경 변수가 있다. UNIX와 Windows의 대표적인 환경 변수로 실행 프로그램의 위치를 지정해 주는 PATH를 들 수 있겠다. Tcl인터프리터는 이러한 환경 변수들을 env라는 전역 변수에 저장하고 있다. 다음을 tclsh에 입력하여 환경 변수가 어떻게 세팅되어 있는지 각자 확인해 보기 바란다.

	#!/usr/local/bin/tclsh

foreach  varName [array names env] {

    puts “$varName = $env($varName)”

}


1.10.3. incr 명령어
incr 명령어는 제어구조는 아니지만 for문에 매우 유용하게 사용되는 명령어이다. incr명령어의 형식은 다음과 같다.

incr variableName ?deltaValue?

deltaValue를 생략하고 사용하면 deltaValue는 기본적으로 1로 설정된다. 명령어가 수행되면 deltaValue값 만큼 variableName이 갖고 있는 값을 증가시킨다.

	%set a 2

2

%incr a

3

%set a

3


deltaValue를 명시하면 명시된 detaValue값 만큼 variabelName의 값을 증가시킨다.

	%set a 2

2

%incr a -1

1

%set a 2

1


단, incr명령어가 취할 수 있는 deltaValue의 값은 정수값에 한정된다.

1.10.4. console 명령어
UNIX의 경우는 기본 환경이 Text 모드이므로 이 명령어가 필요 없으나, Windows나 매킨토시 같은 경우는 기본 환경이 그래픽 유저 인터페이스 이므로 메세지의 출력이나 사용자로 부터 입력을 받을 때는 Text Mode의 console이 필요하다.

특히 Windows와 매킨토시의 경우 console 명령어를 이용하여 console 박스를 보이지 않은 경우에는 puts를 사용하여 메시지를 출력하여도 출력이 되지 않음에 유의 하여야 한다.

UNIX의 경우는 console이란 명령어 자체가 없어서 console이란 명령어를 만나면 invalid command라는 에러가 발생되어 스크립트가 실행되지 않음에 유의한다.

	#!/usr/local/bin/tclsh

# 다음과 같이 콘솔 박스를 보이게 한다.

if { “$tcl_platform(platform)” != “unix” } {

    console show

}

# 다음과 같이 콘솔 박스를 감춘다.

if { “$tcl_platform(platform)” != “unix” } {

    console hide

}


1.10.5. after 명령어
일정한 시간 후에 어떤 명령을 실행하기 위하여 after명령어를 사용한다. after 명령어의 사용 형식은 다음과 같다.

after miliSecond ?command?

10초 간격으로 화면에 Hello를 10번 찍는 프로그램을 다음과 같이 작성할 수 있다. 이 때 사용되는 인터프리터는 tclsh가 아니라 wish임에 유의하도록 하자. tclsh를 사용하지 않고 wish를 사용하는 이유는 적어도 100초 동안 프로그램이 끝나지않고 계속 실행하고 있어야 하기 때문이다.

	#!/usr/local/bin/wish

proc PrintHello { } {

    global count

    puts "Hello"

    incr count

    if { $count < 10 } {

        after 1000 { PrintHello }

    }

}

set count 0

after 1000 { PrintHello }

if { "$tcl_platform(platform)" == "windows" } {

    console show

}


맨 마지막의 command를 생략할 경우 지정한 시간만큼 기다렸다가 다음의 명령을 실행한다. 다음의 예제를 실행하면 Haha가 찍히고 난 후 30초 후에 HoHo가 찍히게 된다.

	#!/usr/local/bin/tclsh

puts Haha

after 3000

puts HoHo




1.10.6. 시스템 변수들
Tcl 인터프리터에 의해 미리 값이 할당되어 있는 시스템 변수들은 다음과 같다.

	변수
	의미

	argc
	command line argument 갯수

	argv
	command line argument를 담고 있는 리스트

	argv0
	실행되고 있는 스크립트의 이름

	env
	환경 변수들의 값을 담고 있는 배열

	tcl_interactive
	스크립트를 실행하고 있으면 0, 

interactive모드이면 1

	tcl_library
	실행 가능한 파일일 경우 1을 리턴한다.

	tcl_patchLevel
	인터프리터의 정확한 버전 ( 패치된 버전 )

	tcl_platform
	OS와 컴퓨터에 관련된 정보를 담고 있는 배열
    tcl_platform(byteOrder) : bigEndian or littleEndian

    tcl_platform(os) : 정확한 OS 이름
    tcl_platform(osVersion) : OS의 버전
    tcl_platform(platform) : OS의 계열 

    tcl_platform(machine) : 컴퓨터 종류

	tcl_pkgPath
	패키지를 찾기 위해 탐색하는 경로를 담은 리스트

	tcl_prompt1
	인터프리터의 프람프트를 사용자가 정의할 때 사용
프람프트를 찍는 데 사용되는 명령어를 할당해야 함

	tcl_prompt2
	인터프리터의 2수준 프람프트를 사용자가 정의할 때
프람프트를 찍는데 사용되는 명령어를 할당해야 함
끝나지 않은 명령어를 여러줄에 걸쳐 입력할 때 

나타나는 프람프트임

	tcl_version
	인터프리터의 버전 정보

	auto_path
	스크립트 라이브러리를 찾는 경로를 담는 리스트

	auto_index
	OS 커맨드를 Tcl 명령어로 aliasing하는데 사용 된다.

예)  % set auto_index(haha) ls

     % haha

	auto_noload
	1로 설정되면 라이브러리 기능을 사용할 수 없다.

	auto_noexec
	1로 설정되면 자동 실행 기능을 사용할 수 없다.

OS 커맨드들을 바로 사용할 수 없고 

exec를 사용해서만 실행 시킬 수 있게 된다.

	geometry
	wish를 사용할 때만 의미가 있으며,

-geometry로 주어진 값을 갖고 있다.

-geometry옵션을 쓰지 않은 경우는 변수 없음


표 12 시스템 변수

1.11. Package

작성하고 있는 프로그램이 작을 경우에는 모든 프로그램에서 사용되는 모든 함수들을 한 파일에 넣어 놓아도 된다. 하지만 규모가 좀 더 커질 경우에는 서로 연관 있는 함수들끼리 모아 따로 다른 파일에 저장해 둔다면 한결 프로그램 소스를 관리하기가 간편해질 것이다. 그리고 자주 사용되는 함수들은 라이브러리화 해서 다른 프로그램에서도 쉽게 사용할 수 있도록 하면 좋을 것이다. Tcl/Tk에서 제공하는 Package 기능을 이용하면 이러한 요구를 쉽게 충족시킬 수 있다.

1.11.1. package 만들기
package를 만들기 위해서는 먼저 package를 구성할 요소들을 만들고 그 다음에 package인덱스를 만들어야 한다. 

package로 사용할 수 있는 것에는 Tcl스크립트와 Tcl C API를 이용해 C언어로 구현된 binary file이 있다. binary파일은 UNIX의 경우 *.so, Windows의 경우 *.dll, 매킨토시의 경우 *.shlib라는 확장자를 가진다. 이들은 모두 각 운영체제에서 제공하는 Dynamic link library의 확장자들이다. 여기서는 Tcl스크립트에 국한하여 설명하도록 한다.

1.11.1.1. package구성 요소 만들기
package를 구성하는 모든 Tcl스크립트 파일은 자신이 속하는 패키지의 이름과 버전을 선언해야 한다. 이를 위해서 사용하는 Tcl명령어는 package provide이다.

package provide packageName MajorVersion.MinorVersion
이렇게 선언된 파일 내에 존재하는 모든 함수들은 같은 패키지에 속하고 그 패키지의 같은 버전에 속하게 된다. 한 패키지를 구성하는 함수들을 여러 개의 파일로 만들 수도 있다. 이 때는 각 파일마다 같은 패키지 이름과 같은 버전을 사용하면 된다.

	#!/usr/local/bin/tclsh

package provide MyPackage 1.0

...


위의 명령이 수행된 다음 줄부터 나오는 함수들은 선언된 패키지에 속하게 된다. 한 파일 내에서 위의 선언을 여러 번 하게 되면 각각의 선언 다음에 나오는 함수들은 각각 다른 패키지에 속하게 된다. 

패키지를 제공하는 파일에는 반드시 서브 루틴들만 들어 있어야 한다. 서브루틴의 내부가 아닌 곳에 존재하는 Tcl명령이 있는 경우에는 원치 않는 결과가 발생하거나 Index파일이 전혀 안만들어 질 수도 있음에 유의해야 한다.

1.11.1.2. package Index만들기
package의 구성 요소들을 적당한 디렉토리에 모아 놓고 pkg_mkIndex라는 명령을 사용하여 package 인덱스를 만든다. 이 인덱스에는 패키지의 종류와 패키지를 구성하는 함수와 파일들에 대한 정보가 저장된다. package를 이용하는 Tcl스크립트를 수행할 때 원하는 함수를 효율적으로 찾기 위한 것이다. 같은 package의 같은 버전을 구성하는 파일들은 될 수 있으면 한 디렉토리 내에 존재하는 것이 좋다. 물론 한 디렉토리 내에 여러 개의 패키지가 존재해도 된다.

예를 들어 “~/tcl/lib/GUI” 디렉토리 밑에 있는 Tcl 스크립트 파일들에 대해 모두 패키지를 만들고자 할 때 tclsh를 실행시키고 다음과 같이 pkg_mkIndex를 실행시킨다.

	% pkg_mkIndex ~/tcl/lib/GUI *.tcl


이 명령을 실행하고 나면 지정한 디렉토리에 pkgIndex.tcl이라는 파일이 생긴다. 이 파일을 열어 보면 지정한 파일에서 package를 구성하는 서브루틴들에 대한 정보가 저장되어 있음을 알 수 있다. 

 pkg_mkIndex 명령어의 사용 형식은 다음과 같다. 첫번째에는 패키지가 존재하는 디렉토리 Path를 줘야하고, 두번째에는 패키지를 구성할 함수들을 담고 있는 파일들에 매칭될 수 있는 패턴을 줘야 한다.

pkg_mkIndex dirPath filePattern ?filePattern ...?

이미 존재하는 package의 내용을 수정할 경우 어떤 경우에 pkg_mkIndex를 실행해야 할까? pkg_mkIndex를 실행해야하는 경우는 새로운 서브루틴을 추가하였거나 이미 존재하고 있던 서브루틴을 삭제한 경우이다. 서브루틴의 추가나 삭제 없이 서브루틴의 내용만 수정한 경우에는 pkg_mkIndex를 해줄 필요가 없다.

1.11.2. package 사용 하기
package를 사용하기 위해서는 먼저 package를 찾기 위해 검색해야 할 디렉토리를 알려줘야하고 어떤 패키지의 어떤 버전을 사용할 것인지 명시해야 한다.

1.11.2.1. 패키지 탐색 경로 지정
Tcl인터프리터는 패키지를 찾기 위해 검색해야 할 디렉토리들을 auto_path라는 변수에 기억하고 있다. 만일 자신이 사용할 패키지가 Tcl설치 시에 기본적으로 제공하는 패키지 디렉토리에 존재하지 않고 다른 곳에 존재한다면, 자신의 패키지 디렉토리를 이 변수에 추가 해야 한다.

auto_path변수는 디렉토리 경로명들을 리스트형태로 보관하고 있으므로 다음과 같이 자신의 패키지 디렉토리를 auto_path에 추가할 수 있다.

	lappend auto_path ~/tcl/lib/GUI


1.11.2.2. 사용할 패키지 선언
package에 대한 인덱스를 만들고 나면 패키지를 사용할 모든 준비는 끝났다. 이제 패키지를 사용하기 위하여 package require라는 Tcl명령어를 사용한다.

package require packageName ?version? ?-exact?

위와 같이 선언하면 스크립트 실행 중 그 파일 안에서 찾을 수 없는 함수들을 선언된 패키지 안에서 찾아 수행 한다. 한 스크립트 내에서 여러 개의 패키지를 사용하는 것도 물론 가능 하다.

패키지의 버전을 명시하지 않으면 같은 이름의 패키지들 중에서 가장 최신 버전의 패키지를 사용한다. 최신 버전은 버전 번호가 가장 큰 것이다. 버전 번호는 먼저 Major Version을 비교하고 Major Version이 같을 경우에만 Minor Version을 비교 한다.

	package require GUI 1.0


버전을 명시했을 때는 정확히 일치하는 버전의 패키지를 사용한다. 만일 정확히 일치하는 버전이 없다면 명시된 Major버전 번호와 같은 Major버전 번호를 갖는 패키지들 중에서 Minor Version번호가 가장 큰 패키지를 사용한다. 하지만 Major버전이 같더라도 명시된 Minor Version번호보다 작은 Minor Version번호를 가진 패키지는 사용할 수 없다. 예를 들어 1.1 버전을 사용하겠다고 했는데 1.1 버전이 없다면, 1.2 버전은 사용할 수 있지만 1.0 버전은 사용할 수 없다는 뜻이다.

-exact옵션이 주어지면 정확히 버전 번호가 일치하는 패키지가 없을 경우 에러를 발생시킨다.

연 습 문 제
앞선 연습문제에서 구현한 Stack, Queue을 Package로 작성하고 이를 사용하는 예를 보이시오.

Stack Level 0





P1의 데이터


	a : 1





Stack Level 1





P2의 데이터


	a : 2





Stack Level 2





P3의 데이터


	a : 3





Stack Level 3





Global 데이터


	a : 4








� exec명령어에 대한 설명은 “� REF _Ref431712835 \n �1.9.1�� REF _Ref431712837 \* MERGEFORMAT �exec�”절에 있다.


� 문법 표시에서 “?”로 싸여진 부분은 생략해도 좋다는 표시이다.


� 삼각함수에서 각도는 라디안 각으로 주어져야 한다.


� 반드시 expr과 함께 사용되어야 한다.


� “� REF _Ref430748595 \n �1.3.2.1�� REF _Ref430748595 \* MERGEFORMAT �문자열 매칭�”에 자세한 설명이 나와 있다.


� “� REF _Ref430452451 \n �1.6�� REF _Ref430452451 \* MERGEFORMAT �정규 표현식�”의 내용을 참조한다.


� 리스트는 “� REF _Ref430745568 \n �1.4�� REF _Ref430745568 \* MERGEFORMAT �리스트(List)�”에서 자세히 설명한다.


� 문자열의 인덱스 값은 0부터 시작한다. 즉, 문자열에서 첫번째 글자의 인덱스는 0이라는 뜻이다.


� “� REF _Ref430748595 \n �1.3.2.1�� REF _Ref430748595 \* MERGEFORMAT �문자열 매칭�” 참조


� “[ ]”의 보다 자세한 사용법은 “� REF _Ref430761881 \n �1.6.1�� REF _Ref430761881 \* MERGEFORMAT �정규 표현식 Premitives�”에 나와 있다.


� “[ ]”의 자세한 사용법은 “� REF _Ref430761881 \n �1.6.1�� PAGEREF _Ref430761881 �58�� REF _Ref430761881 \* MERGEFORMAT �정규 표현식 Premitives�”에 나와 있다.


�명령어 nesting에 관한 설명은 “� REF _Ref430762009 \n �1.1.8�� REF _Ref430762009 \* MERGEFORMAT �명령어의 nesting�”에 있으며, 그루핑과 치환에 관한 설명은 “� REF _Ref430745272 \n �1.1.9�� REF _Ref430745272 \* MERGEFORMAT �그루핑(Grouping)과 변수 치환(Substitution)�” 에 있다.


� Tcl 8.0이전 버전에서 내부적으로 배열은 해쉬 테이블을 이용해 구현되어 인덱스에 상관없이 모든 배열 원소를 참조하는 시간이 같지만, 리스트는 순차적인 검색을 하여 원소를 찾기 때문에 뒤쪽에 있는 원소일 수록 검색 시간이 오래 걸린다.


� “� REF _Ref430748595 \n �1.3.2.1�� REF _Ref430748595 \* MERGEFORMAT �문자열 매칭�”에 자세한 설명이 나와 있다.


� “� REF _Ref430452451 \n �1.6�� REF _Ref430452451 \* MERGEFORMAT �정규 표현식�”을 참조한다.


� Tcl/Tk 8.1버전 이상에서만 이 옵션을 제공한다.


� 이러한 제약은 Tcl/Tk 버전이 올라 가면서 없어질 것이라고 예상되지만 현재는 이러한 제약이 있다.


� 이러한 방식을 call by value라고 한다.


� Tcl 8.0에서는 굳이 array를 명시하지 않고 global만 써줘도 잘 동작한다.


� file join과 file split에 관한 내용은 “� REF _Ref431802893 \n �1.8.3�� REF _Ref431802893 \* MERGEFORMAT �file명령어�”에 나와 있다.


� UNIX의 경우는 csh, bsh 등이 있고 Windows는 command.com이다.


� Tcl에서는 exit명령어를 사용하여 종료 코드 값을 정할 수 있다. C언어에서는 exit()함수를 제공한다.

